
Syntax

Interactive version

gmm (
[
eqname1:

]
<mexp1>) (

[
eqname2:

]
<mexp2>). . .

[
if
] [

in
] [

weight
] [

, options
]

Moment-evaluator program version

gmm moment prog
[

if
] [

in
] [

weight
]
,
{
equations(namelist) | nequations(#)

}
{
parameters(namelist) | nparameters(#)

} [
options

] [
program options

]
where

mexpj is the substitutable expression for the jth moment equation and

moment prog is a moment-evaluator program.

options Description

Model

derivative(<dexpmn>) specify derivative of mexpm with respect to parameter n; can be
specified more than once (interactive version only)

∗twostep use two-step GMM estimator; the default
∗onestep use one-step GMM estimator
∗igmm use iterative GMM estimator

Instruments

instruments(
[
<eqlist>:

]
varlist

[
, noconstant

]
)

specify instruments; can be specified more than once

xtinstruments(
[
<eqlist>:

]
varlist, lags(#1/#2))

specify panel-style instruments; can be specified more than once

Weight matrix

wmatrix(wmtype
[
, independent

]
)

specify weight matrix; wmtype may be robust, cluster clustvar,
hac kernel

[
lags

]
, or unadjusted

center center moments in weight-matrix computation

winitial(iwtype
[
, independent

]
)

specify initial weight matrix; iwtype may be identity,
unadjusted, xt xtspec, or the name of a Stata matrix

2 gmm — Generalized method of moments estimation

Options

variables(varlist) specify variables in model
nocommonesample do not restrict estimation sample to be the same for all equations

SE/Robust

vce(vcetype
[
, independent

]
)

vcetype may be robust, cluster clustvar, bootstrap,
jackknife, hac kernel lags, or unadjusted

quickderivatives use alternative method of computing numerical derivatives
for VCE

Reporting

level(#) set confidence level; default is level(95)

title(string) display string as title above the table of parameter estimates
title2(string) display string as subtitle
display options control column formats and line width

Optimization

from(initial values) specify initial values for parameters
‡igmmiterate(#) specify maximum number of iterations for iterated GMM estimator
‡igmmeps(#) specify # for iterated GMM parameter convergence criterion;

default is igmmeps(1e-6)
‡igmmweps(#) specify # for iterated GMM weight-matrix convergence criterion;

default is igmmweps(1e-6)

optimization options control the optimization process; seldom used

coeflegend display legend instead of statistics

∗ You can specify at most one of these options.
‡ These options may be specified only when igmm is specified.

program options Description

Model

evaluator options additional options to be passed to the moment-evaluator program
∗hasderivatives moment-evaluator program can calculate parameter-level derivatives
∗haslfderivatives moment-evaluator program can calculate linear-form derivatives
†equations(namelist) specify moment-equation names
†nequations(#) specify number of moment equations
‡parameters(namelist) specify parameter names
‡nparameters(#) specify number of parameters

∗ You may not specify both hasderivatives and haslfderivatives.

† You must specify equations(namelist) or nequations(#); you may specify both.

‡ You must specify parameters(namelist) or nparameters(#); you may specify both.

http://www.stata.com/manuals13/u11.pdf#u11.4varlists
http://www.stata.com/manuals13/u12.pdf#u12.4Strings
http://www.stata.com/manuals13/u12.pdf#u12.4Strings


gmm — Generalized method of moments estimation 3

bootstrap, by, jackknife, rolling, statsby, and xi are allowed; see [U] 11.1.10 Prefix commands.
Weights are not allowed with the bootstrap prefix; see [R] bootstrap.
aweights are not allowed with the jackknife prefix; see [R] jackknife.
aweights, fweights, iweights, and pweights are allowed; see [U] 11.1.6 weight.
coeflegend does not appear in the dialog box.
See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

<mexpj> and <dexpmn> are extensions of valid Stata expressions that also contain parameters
to be estimated. The parameters are enclosed in curly braces and must otherwise satisfy the naming
requirements for variables; {beta} is an example of a parameter. Also allowed is a notation of the
form {<eqname>:varlist} for linear combinations of multiple covariates and their parameters. For
example, {xb: mpg price turn} defines a linear combination of the variables mpg, price, and
turn. See Substitutable expressions under Remarks and examples below.

Menu
Statistics > Endogenous covariates > Generalized method of moments estimation

Description
gmm performs generalized method of moments (GMM) estimation. With the interactive version of

the command, you enter the moment equations directly into the dialog box or on the command line
using substitutable expressions. The moment-evaluator program version gives you greater flexibility
in exchange for increased complexity; with this version, you write a program in an ado-file that
calculates the moments based on a vector of parameters passed to it.

gmm can fit both single- and multiple-equation models, and it allows moment conditions of the
form E{ziui(β)} = 0, where zi is a vector of instruments and ui(β) is often an additive regression
error term, as well as more general moment conditions of the form E{hi(zi;β)} = 0. gmm works
with cross-sectional, time-series, and longitudinal (panel) data.

Options

� � �
Model �

derivative(
[

eqname | #
]
/name = <dexpmn>) specifies the derivative of moment equation eqname

or # with respect to parameter name. If eqname or # is not specified, gmm assumes that the derivative
applies to the first moment equation.

For a moment equation of the form E{zmiumi(β)} = 0, derivative(m/βj = <dexpmn>) is
to contain a substitutable expression for ∂umi/∂βj .

For a moment equation of the form E{hmi(zi;β)} = 0, derivative(m/βj = <dexpmn>) is
to contain a substitutable expression for ∂hmi/∂βj .

<dexpmn> uses the same substitutable expression syntax as is used to specify moment equations.
If you declare a linear combination in a moment equation, you provide the derivative for the linear
combination; gmm then applies the chain rule for you. See Specifying derivatives under Remarks
and examples below for examples.

If you do not specify the derivative() option, gmm calculates derivatives numerically. You must
either specify no derivatives or specify all the derivatives that are not identically zero; you cannot
specify some analytic derivatives and have gmm compute the rest numerically.

http://www.stata.com/manuals13/u11.pdf#u11.1.10Prefixcommands
http://www.stata.com/manuals13/rbootstrap.pdf#rbootstrap
http://www.stata.com/manuals13/rjackknife.pdf#rjackknife
http://www.stata.com/manuals13/u11.pdf#u11.1.6weight
http://www.stata.com/manuals13/u20.pdf#u20Estimationandpostestimationcommands


4 gmm — Generalized method of moments estimation

twostep, onestep, and igmm specify which estimator is to be used. You can specify at most one
of these options. twostep is the default.

twostep requests the two-step GMM estimator. gmm obtains parameter estimates based on the initial
weight matrix, computes a new weight matrix based on those estimates, and then reestimates the
parameters based on that weight matrix.

onestep requests the one-step GMM estimator. The parameters are estimated based on an initial
weight matrix, and no updating of the weight matrix is performed except when calculating the
appropriate variance–covariance (VCE) matrix.

igmm requests the iterative GMM estimator. gmm obtains parameter estimates based on the initial
weight matrix, computes a new weight matrix based on those estimates, reestimates the parameters
based on that weight matrix, computes a new weight matrix, and so on, to convergence. Convergence
is declared when the relative change in the parameter vector is less than igmmeps(), the relative
change in the weight matrix is less than igmmweps(), or igmmiterate() iterations have been
completed. Hall (2005, sec. 2.4 and 3.6) mentions that there may be gains to finite-sample efficiency
from using the iterative estimator.

� � �
Instruments �

instruments(
[
<eqlist>:

]
varlist

[
, noconstant

]
) specifies a list of instrumental variables to be

used. If you specify a single moment equation, then you do not need to specify the equations to
which the instruments apply; you can omit the eqlist and simply specify instruments(varlist).
By default, a constant term is included in varlist; to omit the constant term, use the noconstant
suboption: instruments(varlist, noconstant).

If you specify a model with multiple moment conditions of the form

E

 z1iu1i(β)
· · ·

zqiuqi(β)

 = 0

then you can specify the equations to indicate the moment equations for which the list of variables
is to be used as instruments if you do not want that list applied to all the moment equations. For
example, you might type

gmm (main:<mexp1>) (<mexp2>) (<mexp3>), instruments(z1 z2) ///
instruments(2: z3) instruments(main 3: z4)

Variables z1 and z2 will be used as instruments for all three equations, z3 will be used as an
instrument for the second equation, and z4 will be used as an instrument for the first and third
equations. Notice that we chose to supply a name for the first moment equation but not the second
two.

varlist may contain factor variables and time-series operators; see [U] 11.4.3 Factor variables and
[U] 11.4.4 Time-series varlists, respectively.

xtinstruments(
[
<eqlist>:

]
varlist, lags(#1/#2)) is for use with panel-data models in which the

set of available instruments depends on the time period. As with instruments(), you can prefix
the list of variables with equation names or numbers to target instruments to specific equations.
Unlike with instruments(), a constant term is not included in varlist. You must xtset your
data before using this option; see [XT] xtset.

If you specify

gmm . . ., xtinstruments(x, lags(1/.)) . . .

http://www.stata.com/manuals13/u11.pdf#u11.4varlists
http://www.stata.com/manuals13/u11.pdf#u11.4.3Factorvariables
http://www.stata.com/manuals13/u11.pdf#u11.4.4Time-seriesvarlists
http://www.stata.com/manuals13/u11.pdf#u11.4varlists
http://www.stata.com/manuals13/xtxtset.pdf#xtxtset

then for panel i and period t, gmm uses as instruments xi,t−1, xi,t−2, . . . , xi1. More generally,
specifying xtinstruments(x, lags(#1, #2)) uses as instruments xi,t−#1 , . . . , xi,t−#2 ; setting
#2 = . requests all available lags. #1 and #2 must be zero or positive integers.

gmm automatically excludes observations for which no valid instruments are available. It does,
however, include observations for which only a subset of the lags is available. For example, if you
request that lags one through three be used, then gmm will include the observations for the second
and third time periods even though fewer than three lags are available as instruments.

� � �
Weight matrix �

wmatrix(wmtype
[
, independent

]
) specifies the type of weight matrix to be used in conjunction

with the two-step and iterated GMM estimators.

Specifying wmatrix(robust) requests a weight matrix that is appropriate when the errors are
independent but not necessarily identically distributed. wmatrix(robust) is the default.

Specifying wmatrix(cluster clustvar) requests a weight matrix that accounts for arbitrary
correlation among observations within clusters identified by clustvar.

Specifying wmatrix(hac kernel #) requests a heteroskedasticity- and autocorrelation-consistent
(HAC) weight matrix using the specified kernel (see below) with # lags. The bandwidth of a kernel
is equal to the number of lags plus one.

Specifying wmatrix(hac kernel opt) requests an HAC weight matrix using the specified kernel,
and the lag order is selected using Newey and West’s (1994) optimal lag-selection algorithm.

Specifying wmatrix(hac kernel) requests an HAC weight matrix using the specified kernel and
N − 2 lags, where N is the sample size.

There are three kernels available for HAC weight matrices, and you may request each one by using
the name used by statisticians or the name perhaps more familiar to economists:

bartlett or nwest requests the Bartlett (Newey–West) kernel;

parzen or gallant requests the Parzen (Gallant) kernel; and

quadraticspectral or andrews requests the quadratic spectral (Andrews) kernel.

Specifying wmatrix(unadjusted) requests a weight matrix that is suitable when the errors are
homoskedastic. In some applications, the GMM estimator so constructed is known as the (nonlinear)
two-stage least-squares (2SLS) estimator.

Including the independent suboption creates a weight matrix that assumes moment equations are
independent. This suboption is often used to replicate other models that can be motivated outside
the GMM framework, such as the estimation of a system of equations by system-wide 2SLS. This
suboption has no effect if only one moment equation is specified.

wmatrix() has no effect if onestep is also specified.

center requests that the sample moments be centered (demeaned) when computing GMM weight
matrices. By default, centering is not done.

winitial(wmtype
[
, independent

]
) specifies the weight matrix to use to obtain the first-step

parameter estimates.

Specifying winitial(unadjusted) requests a weight matrix that assumes the moment equations
are independent and identically distributed. This matrix is of the form (Z′Z)−1, where Z represents
all the instruments specified in the instruments() option. To avoid a singular weight matrix,
you should specify at least q − 1 moment equations of the form E{zhiuhi(β)} = 0, where q is
the number of moment equations, or you should specify the independent suboption.

Including the independent suboption creates a weight matrix that assumes moment equations are
independent. Elements of the weight matrix corresponding to covariances between two moment
equations are set equal to zero. This suboption has no effect if only one moment equation is
specified.

winitial(unadjusted) is the default.

winitial(xt xtspec) is for use with dynamic panel-data models in which one of the moment
equations is specified in first-differences form. xtspec is a string consisting of the letters “L” and
“D”, the length of which is equal to the number of moment equations in the model. You specify
“L” for a moment equation if that moment equation is written in levels, and you specify “D” for a
moment equation if it is written in first-differences; xtspec is not case sensitive. When you specify
this option, you can specify at most one moment equation in levels and one moment equation
in first-differences. See the examples listed in Dynamic panel-data models under Remarks and
examples below.

winitial(identity) requests that the identity matrix be used.

winitial(matname) requests that Stata matrix matname be used. You cannot specify the inde-
pendent suboption if you specify winitial(matname).

� � �
Options �

variables(varlist) specifies the variables in the model. gmm ignores observations for which any of
these variables has a missing value. If you do not specify variables(), then gmm assumes all the
observations are valid and issues an error message with return code 480 if any moment equations
evaluate to missing for any observations at the initial value of the parameter vector.

nocommonesample requests that gmm not restrict the estimation sample to be the same for all
equations. By default, gmm will restrict the estimation sample to observations that are available
for all equations in the model, mirroring the behavior of other multiple-equation estimators such
as nlsur, sureg, or reg3. For certain models, however, different equations can have different
numbers of observations. For these models, you should specify nocommonesample. See Dynamic
panel-data models below for one application of this option. You cannot specify weights if you
specify nocommonesample.

� � �
SE/Robust �

vce(vcetype
[
, independent

]
) specifies the type of standard error reported, which includes types

that are robust to some kinds of misspecification (robust), that allow for intragroup correlation
(cluster clustvar), and that use bootstrap or jackknife methods (bootstrap, jackknife); see
[R] vce option.

vce(unadjusted) specifies that an unadjusted (nonrobust) VCE matrix be used; this, along with
the twostep option, results in the “optimal two-step GMM” estimates often discussed in textbooks.

The default vcetype is based on the wmtype specified in the wmatrix() option. If wmatrix()
is specified but vce() is not, then vcetype is set equal to wmtype. To override this behavior and
obtain an unadjusted (nonrobust) VCE matrix, specify vce(unadjusted).

Specifying vce(bootstrap) or vce(jackknife) results in standard errors based on the bootstrap
or jackknife, respectively. See [R] vce option, [R] bootstrap, and [R] jackknife for more information
on these VCEs.

The syntax for vcetypes other than bootstrap and jackknife is identical to those for wmatrix().

http://www.stata.com/manuals13/u11.pdf#u11.4varlists
http://www.stata.com/manuals13/rvce_option.pdf#rvce_option
http://www.stata.com/manuals13/rvce_option.pdf#rvce_option
http://www.stata.com/manuals13/rbootstrap.pdf#rbootstrap
http://www.stata.com/manuals13/rjackknife.pdf#rjackknife

quickderivatives requests that an alternative method be used to compute the numerical derivatives
for the VCE. This option has no effect if you specify the derivatives(), hasderivatives, or
haslfderivatives option.

The VCE depends on a matrix of partial derivatives that gmm must compute numerically unless you
supply analytic derivatives. This Jacobian matrix will be especially large if your model has many
instruments, moment equations, or parameters.

By default, gmm computes each element of the Jacobian matrix individually, searching for an optimal
step size each time. Although this procedure results in accurate derivatives, it is computationally
taxing: gmm may have to evaluate the moments of your model five or more times for each element
of the Jacobian matrix.

When you specify the quickderivatives option, gmm computes all derivatives corresponding to
a parameter at once, using a fixed step size proportional to the parameter’s value. This method
requires just two evaluations of the model’s moments to compute an entire column of the Jacobian
matrix and therefore has the most impact when you specify many instruments or moment equations.

Most of the time, the two methods produce virtually identical results, but the quickderivatives
method may fail if a moment equation is highly nonlinear or if instruments differ by orders of
magnitude. In the rare case where you specify quickderivatives and obtain suspiciously large
or small standard errors, try refitting your model without this option.

� � �
Reporting �

level(#); see [R] estimation options.

title(string) specifies an optional title that will be displayed just above the table of parameter
estimates.

title2(string) specifies an optional subtitle that will be displayed between the title specified in
title() and the table of parameter estimates. If title2() is specified but title() is not,
title2() has the same effect as title().

display options: cformat(% fmt), pformat(% fmt), sformat(% fmt), and nolstretch; see [R] es-
timation options.

� � �
Optimization �

from(initial values) specifies the initial values to begin the estimation. You can specify a 1 × k
matrix, where k is the number of parameters in the model, or you can specify a parameter name,
its initial value, another parameter name, its initial value, and so on. For example, to initialize
alpha to 1.23 and delta to 4.57, you would type

gmm ..., from(alpha 1.23 delta 4.57) ...

Initial values declared using this option override any that are declared within substitutable expres-
sions. If you specify a parameter that does not appear in your model, gmm exits with error code
480. If you specify a matrix, the values must be in the same order in which the parameters are
declared in your model. gmm ignores the row and column names of the matrix.

igmmiterate(#), igmmeps(#), and igmmweps(#) control the iterative process for the iterative
GMM estimator. These options can be specified only if you also specify igmm.

igmmiterate(#) specifies the maximum number of iterations to perform with the iterative GMM
estimator. The default is the number set using set maxiter (set [R] maximize), which is
16,000 by default.

http://www.stata.com/manuals13/restimationoptions.pdf#restimationoptions
http://www.stata.com/manuals13/u12.pdf#u12.4Strings
http://www.stata.com/manuals13/u12.pdf#u12.4Strings
http://www.stata.com/manuals13/d.pdf#dformat
http://www.stata.com/manuals13/restimationoptions.pdf#restimationoptions
http://www.stata.com/manuals13/restimationoptions.pdf#restimationoptions
http://www.stata.com/manuals13/rmaximize.pdf#rmaximize

igmmeps(#) specifies the convergence criterion used for successive parameter estimates when the
iterative GMM estimator is used. The default is igmmeps(1e-6). Convergence is declared when
the relative difference between successive parameter estimates is less than igmmeps() and the
relative difference between successive estimates of the weight matrix is less than igmmweps().

igmmweps(#) specifies the convergence criterion used for successive estimates of the weight matrix
when the iterative GMM estimator is used. The default is igmmweps(1e-6). Convergence is
declared when the relative difference between successive parameter estimates is less than
igmmeps() and the relative difference between successive estimates of the weight matrix is
less than igmmweps().

optimization options: technique(), conv maxiter(), conv ptol(), conv vtol(),
conv nrtol(), tracelevel(). technique() specifies the optimization technique to use; gn
(the default), nr, dfp, and bfgs are allowed. conv maxiter() specifies the maximum number
of iterations; conv ptol(), conv vtol(), and conv nrtol() specify the convergence criteria
for the parameters, gradient, and scaled Hessian, respectively. tracelevel() allows you to obtain
additional details during the iterative process. See [M-5] optimize( ).

The following options pertain only to the moment-evaluator program version of gmm.

� � �
Model �

evaluator options refer to any options allowed by your moment prog.

hasderivatives and haslfderivatives indicate that you have written your moment-evaluator
program to compute derivatives. You may specify one or the other but not both. If you do not
specify either of these options, gmm computes the derivatives numerically.

hasderivatives indicates that your moment-evaluator program computes parameter-level deriva-
tives.

haslfderivatives indicates that your moment-evaluator program computes equation-level
derivatives and is useful only when you specify the parameters of your model using the
<eqname>:<varname> syntax of the parameters() option.

See Details of moment-evaluator programs below for more information.

equations(namelist) specifies the names of the moment equations in the model. If you specify both
equations() and nequations(), the number of names in the former must match the number
specified in the latter.

nequations(#) specifies the number of moment equations in the model. If you do not specify
names with the equations() option, gmm numbers the moment equations 1, 2, 3, . . . . If you
specify both equations() and nequations(), the number of names in the former must match
the number specified in the latter.

parameters(namelist) specifies the names of the parameters in the model. The names of the
parameters must adhere to the naming conventions of Stata’s variables;
see [U] 11.3 Naming conventions.

Alternatively, you may specify a list of names in which each item in the list is of the form
<eqname>:<varname>, where eqname is an equation name used to group parameters, and
varname is the name of an existing variable or cons to indicate a constant term. When you
use this syntax, gmm adorns the parameter vector passed to your evaluator program with these
names so that you can use matrix score (see [P] matrix score) to compute linear combinations
of parameters. These equation names are not related to the names you may give to the moment
equations.

http://www.stata.com/manuals13/m-5optimize.pdf#m-5optimize()
http://www.stata.com/manuals13/u11.pdf#u11.3Namingconventions
http://www.stata.com/manuals13/pmatrixscore.pdf#pmatrixscore

If you specify both parameters() and nparameters(), the number of names in the former
must match the number specified in the latter.

nparameters(#) specifies the number of parameters in the model. If you do not specify names with
the parameters() option, gmm names them b1, b2, . . . , b#. If you specify both parameters()
and nparameters(), the number of names in the former must match the number specified in the
latter.

The following option is available with gmm but is not shown in the dialog box:

coeflegend; see [R] estimation options.

Remarks and examples stata.com

Remarks are presented under the following headings:

Introduction
Substitutable expressions
The weight matrix and two-step estimation
Obtaining standard errors
Exponential (Poisson) regression models
Specifying derivatives
Exponential regression models with panel data
Rational-expectations models
System estimators
Dynamic panel-data models
Details of moment-evaluator programs

Introduction

The generalized method of moments (GMM) estimator is a workhorse of modern econometrics and
is discussed in all the leading textbooks, including Cameron and Trivedi (2005, 2010), Davidson and
MacKinnon (1993, 2004), Greene (2012, 468–506), Ruud (2000), Hayashi (2000), Wooldridge (2010),
Hamilton (1994), and Baum (2006). An excellent treatise on GMM with a focus on time-series
applications is Hall (2005). The collection of papers by Mátyás (1999) provides both theoretical and
applied aspects of GMM. Here we give a brief introduction to the methodology and emphasize how
the various options of gmm are used.

The starting point for the generalized method of moments (GMM) estimator is the analogy principle,
which says we can estimate a parameter by replacing a population moment condition with its sample
analogue. For example, the mean of an independent and identically distributed (i.i.d.) population is
defined as the value µ such that the first (central) population moment is zero; that is, µ solves
E(y− µ) = 0 where y is a random draw from the population. The analogy principle tells us that to
obtain an estimate, µ̂, of µ, we replace the population-expectations operator with its sample analogue
(Manski 1988; Wooldridge 2010):

E(y − µ) = 0 −→ 1

N

N∑
i=1

(yi − µ̂) = 0 −→ µ̂ =
1

N

N∑
i=1

yi

where N denotes sample size and yi represents the ith observation of y in our dataset. The estimator
µ̂ is known as the method of moments (MM) estimator, because we started with a population moment
condition and then applied the analogy principle to obtain an estimator that depends on the observed
data.

http://www.stata.com/manuals13/restimationoptions.pdf#restimationoptions
http://stata.com

Ordinary least-squares (OLS) regression can also be viewed as an MM estimator. In the model

y = x′β + u

we assume that u has mean zero conditional on x: E(u|x) = 0. This conditional expectation implies
the unconditional expectation E(xu) = 0 because, using the law of iterated expectations,

E(xu) = Ex {E(xu|x)} = Ex {xE(u|x)} = 0

(Using the law of iterated expectations to derive unconditional expectations based on conditional
expectations, perhaps motivated by subject theory, is extremely common in GMM estimation.) Con-
tinuing,

E(xu) = E {x(y − x′β)} = 0

Applying the analogy principle,

E {x(y − x′β)} −→ 1

N

N∑
i=1

xi(yi − x′iβ) = 0

so that

β̂ =
(∑

i
xix
′
i

)−1∑
i
xiyi

which is just the more familiar formula β̂ = (X′X)
−1

X′y written using summation notation.

In both the previous examples, the number of parameters we were estimating equaled the number
of moment conditions. In the first example, we estimated one parameter, µ, and had one moment
condition E(y − µ) = 0. In the second example, the parameter vector β had k elements, as did
the vector of regressors x, yielding k moment conditions. Ignoring peculiar cases, a model of m
equations in m unknowns has a unique solution, and because the moment equations in these examples
were linear, we were able to solve for the parameters analytically. Had the moment conditions been
nonlinear, we would have had to use numerical techniques to solve for the parameters, but that is not
a significant limitation with modern computers.

What if we have more moment conditions than parameters? Say we have l moment conditions
and k parameters. A model of l > k equations in k unknowns does not have a unique solution.
Any size-k subset of the moment conditions would yield a consistent parameter estimate, though the
parameter estimate so obtained would in general be different based on which k moment conditions
we used.

For concreteness, let’s return to our regression model,

y = x′β + u

but we no longer wish to assume that E(xu) = 0; we suspect that the error term u affects one
or more elements of x. As a result, we can no longer use the OLS estimator. Suppose we have a
vector z with the properties that E(zu) = 0, that the rank of E(z′z) equals l, and that the rank
of E(z′x) = k. The first assumption simply states that z is not correlated with the error term. The
second assumption rules out perfect collinearity among the elements of z. The third assumption,
known as the rank condition in econometrics, ensures that z is sufficiently correlated with x and
that the estimator is feasible. If some elements of x are not correlated with u, then they should also
appear in z.

If l < k, then the rank of E(z′x) < k, violating the rank condition.

If l = k, then we can use the simpler MM estimator we already discussed; we would obtain what
is sometimes called the simple instrumental-variables estimator β̂ = (

∑
i zix

′
i)
−1∑

i ziyi. The rank
condition ensures that

∑
i zix

′
i is invertible, at least in the population.

If l > k, the GMM estimator chooses the value, β̂, that minimizes a quadratic function of the
moment conditions. We could define

β̂ ≡ arg minβ

{
1

N

∑
i
ziui(β)

}′{
1

N

∑
i
ziui(β)

}
(1)

where for our linear regression example ui(β) = yi−x′iβ. This estimator tries to make the moment
conditions as close to zero as possible. This simple estimator, however, applies equal weight to each
of the moment conditions; and as we shall see later, we can obtain more efficient estimators by
choosing to weight some moment conditions more highly than others.

Consider the quadratic function

Q(β) =

{
1

N

∑
i
ziui(β)

}′
W

{
1

N

∑
i
ziui(β)

}
where W is a symmetric positive-definite matrix known as a weight matrix. Then we define the GMM
estimator as

β̂ ≡ arg minβ Q(β) (2)

Continuing with our regression model example, if we choose

W =

(
1

N

∑
i
ziz
′
i

)−1
(3)

then we obtain

β̂ =

{(
1

N

∑
i
xiz
′
i

)(
1

N

∑
i
ziz
′
i

)−1(
1

N

∑
i
zix
′
i

)}−1
×

(
1

N

∑
i
xiz
′
i

)(
1

N

∑
i
ziz
′
i

)−1(
1

N

∑
i
ziyi

)
which is the well-known two-stage least-squares (2SLS) estimator. Our choice of weight matrix here
was based on the assumption that u was homoskedastic. A feature of GMM estimation is that by
selecting different weight matrices, we can obtain estimators that can tolerate heteroskedasticity,
clustering, autocorrelation, and other features of u. See [R] ivregress for more information about the
2SLS and linear GMM estimators.

Returning to the case where the model is “just identified”, meaning that l = k, if we apply the
GMM estimator, we will obtain the same estimate, β̂, regardless of our choice of W. Because l = k,
if a unique solution exists, it will set all the sample moment conditions to zero jointly, so W has no
impact on the value of β that minimizes the objective function.

We will highlight other features of the GMM estimator and the gmm command as we proceed
through examples. First, though, we discuss how to specify moment equations by using substitutable
expressions.

http://www.stata.com/manuals13/rivregress.pdf#rivregress


12 gmm — Generalized method of moments estimation

Substitutable expressions

To use the interactive version of gmm, you define the moment equations by using substitutable
expressions. In most applications, your moment conditions are of the form E {ziui(β)}, where ui(β)
is a residual term that depends on the parameter vector β as well as variables in your dataset, though
we suppress expressing the variables for notational simplicity; we refer to ui(β) as the moment
equation to differentiate it from the moment conditions E{z′iui(β)} = 0.

Substitutable expressions in gmm work much like those used in nl and nlsur, though with one
important difference. For the latter two commands, you type the name of the dependent variable,
an equal sign, and then the regression function. For example, in nl, if you want to fit the function
y = f(x;β) + u, you would type

nl (y = <expression for f(x; β)>), ...

On the other hand, gmm requires you to write a substitutable expression for u; in this example,
u = y − f(x;β), so you would type

gmm (y - <expression for f(x; β)>), ...

The advantage of writing the substitutable expression directly in terms of u is that you are not
restricted to fitting models with additive error terms as you are with nl and nlsur.

You specify substitutable expressions just like any other mathematical expression involving scalars
and variables, such as those you would use with Stata’s generate command, except that the
parameters to be estimated are bound in braces. See [U] 13.2 Operators and [U] 13.3 Functions
for more information on expressions. Parameter names must follow the same conventions as variable
names. See [U] 11.3 Naming conventions.

For example, say that the tth observation on a sample moment is

ut = 1− β
{

(1 + rt+1)(ct+1/ct)
−γ}

where t denotes time period, β and γ are the parameters to be estimated, and r and c are variables
in your dataset. Then you would type

gmm (1 - {beta}*((1 + F.r)*(F.c/c)^(-1*{gamma}))), ...

Because β and γ are parameters, we enclose them in braces. Also notice our use of the forward
operator to refer to the values of r and c one period ahead; time-series operators are allowed
in substitutable expressions as long as you have previously tsset (see [TS] tsset) your data. See
[U] 13.9 Time-series operators for more information on time-series operators.

To specify initial values for some parameters, you can include an equal sign and the initial value
after a parameter:

gmm (1 - {beta}*((1 + F.r)*(F.c/c)^(-1*{gamma=1}))), ...

would initialize γ to be one. If you do not specify an initial value for a parameter, it is initialized to
zero.

Frequently, even nonlinear functions contain linear combinations of variables. As an example,
suppose you have this moment equation:

u = {y − exp(β1x1 + β2x2 + β3x3)} /exp(β1x1 + β2x2 + β3x3)

Instead of typing

gmm ((y - exp({beta1}*x1 + {beta2}*x2 + {beta3}*x3)) / ///
exp({beta1}*x1 + {beta2}*x2 + {beta3}*x3)) ...

http://www.stata.com/manuals13/u13.pdf#u13.2Operators
http://www.stata.com/manuals13/u13.pdf#u13.3Functions
http://www.stata.com/manuals13/u11.pdf#u11.3Namingconventions
http://www.stata.com/manuals13/tstsset.pdf#tstsset
http://www.stata.com/manuals13/u13.pdf#u13.9Time-seriesoperators

you can type

gmm ((y - exp({xb:x1 x2 x3})) / exp({xb:})) .....

The notation {xb:x1 x2 x3} tells gmm that you want a linear combination of the variables x1, x2,
and x3. We named this linear combination xb, so gmm will name the three parameters corresponding
to the three variables xb x1, xb x2, and xb x3. You can name the linear combination anything
you wish (subject to Stata’s naming conventions for variable names); gmm then names the parameter
corresponding to variable x lc x, where lc is the name of your linear combination. You cannot use
the same name for both an individual parameter and a linear combination. You can, however, refer to
one parameter in a linear combination after it has been declared as you would any other parameter
by using the notation {lc x}. Linear combinations do not include a constant term.

Once we have declared the variables in the linear combination xb, we can subsequently refer
to the linear combination in our substitutable expression by using the notation xb:. The colon is
not optional; it tells gmm that you are referring to a previously declared linear combination, not an
individual parameter. This shorthand notation is also handy when specifying derivatives, as we will
show later.

In general, there are three rules to follow when defining substitutable expressions:

1. Parameters of the model are bound in braces: {b0}, {param}, etc.

2. Initial values for parameters are given by including an equal sign and the initial value inside
the braces: {b0=1}, {param=3.571}, etc.

3. Linear combinations of variables can be included using the notation {eqname:varlist}: {xb:
mpg price weight}, {score: w x z}, etc. Parameters of linear combinations are initialized
to zero.

If you specify initial values by using the from() option, they override whatever initial values
are given within the substitutable expression. Substitutable expressions are so named because, once
values are assigned to the parameters, the resulting expressions can be handled by generate and
replace.

Example 1: OLS regression

In Introduction, we stated that OLS is an MM estimator. Say that we want to fit the model

mpg = β0 + β1weight + β2length + u

where u is an i.i.d. error term. We type

. use http://www.stata-press.com/data/r13/auto
(1978 Automobile Data)

. gmm (mpg - {b1}*weight - {b2}*length - {b0}), instruments(weight length)

Step 1
Iteration 0: GMM criterion Q(b) = 475.4138
Iteration 1: GMM criterion Q(b) = 3.305e-20
Iteration 2: GMM criterion Q(b) = 3.795e-27

Step 2
Iteration 0: GMM criterion Q(b) = 7.401e-28
Iteration 1: GMM criterion Q(b) = 3.771e-31

GMM estimation

Number of parameters = 3
Number of moments = 3
Initial weight matrix: Unadjusted Number of obs = 74
GMM weight matrix: Robust

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/b1 -.0038515 .0019472 -1.98 0.048 -.0076678 -.0000351
/b2 -.0795935 .0677528 -1.17 0.240 -.2123864 .0531995
/b0 47.88487 7.505985 6.38 0.000 33.17341 62.59633

Instruments for equation 1: weight length _cons

Recall that the moment condition for OLS regression is E(xu) = 0, where x, the list of instruments, is
the same as the list of regressors in the model. In our command, we defined the residual term, u, inside
parentheses by using a substitutable expression; because linear combinations declared in substitutable
expressions do not include a constant term, we included our own (b0). Inside the instruments()
option, we listed our instruments; by default, gmm includes a constant term among the instrument list.

Because the number of moments equals the number of parameters we are estimating, the model
is said to be “just identified” or “exactly identified.” Therefore, the choice of weight matrix has no
impact on the solution to (2), and the criterion function Q(β) achieves its minimum value at zero.

The OLS estimator is a one-step GMM estimator, but we did not bother to specify the onestep
option because the model is just identified. Doing a second step of GMM estimation affects neither
the point estimates nor the standard errors, so to keep the syntax as simple as possible, we did not
include the onestep option. The first step of estimation resulted in Q(β) = 0 as expected, and the
second step of estimation did not change the minimized value of Q(β). (4× 10−27 and 3× 10−31

are both zero for all practical purposes.)

When you do not specify either the wmatrix() or the vce() option, gmm reports heteroskedasticity-
robust standard errors. The parameter estimates reported here match those that we would obtain from
the command

. regress mpg weight length, vce(robust)

The standard errors reported by that regress command would be larger than those reported by gmm by
a factor of sqrt(74/71) because regress makes a small-sample adjustment to the estimated variance
matrix while gmm does not. Likewise, had we specified the vce(unadjusted) option with our gmm
command, then our standard errors would differ by a factor of sqrt(74/71) from those reported by
regress without the vce(robust) option.

Using the notation for linear combinations of parameters, we could have typed

. gmm (mpg - {xb: weight length} - {b0}), instruments(weight length)

and obtained identical results. Instead of having parameters b1 and b2, with this syntax we would
have parameters xb weight and xb length.

Example 2: Instrumental-variables regression

In Introduction, we mentioned that 2SLS can be viewed as a GMM estimator. In example 1 of
[R] ivregress, we fit by 2SLS a model of rental rates (rent) as a function of the value of owner-occupied
housing (hsngval) and the percentage of the population living in urban areas (pcturban):

rent = β0 + β1hsngval + β2pcturban + u

by 2SLS. We argued that random shocks that affect rental rates likely also affect housing values, so
we treated hsngval as an endogenous variable. As additional instruments, we used family income,
faminc, and three regional dummies (reg2–reg4).

To replicate the results of ivregress 2sls by using gmm, we type
. use http://www.stata-press.com/data/r13/hsng2
(1980 Census housing data)

. gmm (rent - {xb:hsngval pcturban} - {b0}),
> instruments(pcturban faminc reg2-reg4) vce(unadjusted) onestep

Step 1
Iteration 0: GMM criterion Q(b) = 56115.03
Iteration 1: GMM criterion Q(b) = 110.91583
Iteration 2: GMM criterion Q(b) = 110.91583

GMM estimation

Number of parameters = 3
Number of moments = 6
Initial weight matrix: Unadjusted Number of obs = 50

Coef. Std. Err. z P>|z| [95% Conf. Interval]

/xb_hsngval .0022398 .0003284 6.82 0.000 .0015961 .0028836
/xb_pcturban .081516 .2987652 0.27 0.785 -.504053 .667085

/b0 120.7065 15.22839 7.93 0.000 90.85942 150.5536

Instruments for equation 1: pcturban faminc reg2 reg3 reg4 _cons

We specified vce(unadjusted) so that we would obtain an unadjusted VCE matrix and our
standard errors would match those reported in [R] ivregress.

Pay attention to how we specified the instruments() option. In Introduction, we mentioned
that the moment conditions for the 2SLS estimator are E(zu) = 0, and we mentioned that if some
elements of x (the regressors) are not endogenous, then they should also appear in z. In this model,
we assume the regressor pcturban is exogenous, so we included it in the list of instrumental
variables. Commands like ivregress, ivprobit, and ivtobit accept standard varlists, so they can
deduce the exogenous regressors in the model. Because gmm accepts arbitrary functions in the form
of substitutable expressions, it has no way of discerning the exogenous variables of the model on its
own.

Also notice that we specified the onestep option. The 2SLS estimator is a one-step GMM estimator
that is based on a weight matrix that assumes the error terms are i.i.d. Unlike the previous example,
here we had more instruments than parameters, so the minimized value of Q(β) is nonzero. We
discuss the weight matrix and its relationship to two-step estimation next.

http://www.stata.com/manuals13/rivregress.pdf#rivregressRemarksandexamplesex_ivregress_2sls
http://www.stata.com/manuals13/rivregress.pdf#rivregress
http://www.stata.com/manuals13/rivregress.pdf#rivregress

The weight matrix and two-step estimation

Recall our definition of the GMM estimator given in (2). The estimator, β̂, depends on the choice
of the weight matrix, W. Under relatively mild assumptions, our estimator, β̂, is consistent regardless
of the choice of W, so how are we to decide what W to use? The most common solution is to use
the two-step estimator, which we now describe.

A key result in Hansen’s (1982) seminal paper is that if we denote by S the covariance matrix of
the moment conditions, then the optimal (in a way we make precise later) GMM estimator is the one
that uses a weight matrix equal to the inverse of the moment covariance matrix. That is, if we let
S = Cov(zu), then we want to use W = S−1. But how do we obtain S in the first place?

If we assume that the errors are i.i.d., then

Cov(zu) = E(u2zz′) = σ2E(zz′)

where σ2 is the variance of u. Because σ2 is a positive scalar, we can ignore it when solving (2).
Thus we compute

Ŵ1 =

(
1

N

∑
i
ziz
′
i

)−1
(4)

which does not depend on any unknown model parameters. (Notice that Ŵ1 is the same weight
matrix used in 2SLS.) Given Ŵ1, we can solve (2) to obtain an initial estimate, say, β̂1.

Our estimate, β̂1, is consistent, so by Slutsky’s theorem, the sample residuals û computed at
this value of β will also be consistent. Using virtually the same arguments used to justify the
Huber/Eicker/White heteroskedasticity-robust VCE, if we assume that the residuals are independent
though not identically distributed, we can estimate S as

Ŝ =
1

N

∑
i

û2i ziz
′
i

Then, in the second step, we re-solve (2), using Ŵ2 = Ŝ−1, yielding the two-step GMM estimate
β̂2. If the residuals exhibit clustering, you can specify wmatrix(cluster varname) so that gmm
computes a weight matrix that does not assume the ui’s are independent within clusters identified by
varname. You can specify wmatrix(hac . . .) to obtain weight matrices that are suitable for when
the ui’s exhibit autocorrelation as well as heteroskedasticity.

We could take the point estimates from the second round of estimation and use them to compute
yet another weight matrix, Ŵ3, say, re-solve (2) yet again, and so on, stopping when the parameters
or weight matrix do not change much from one iteration to the next. This procedure is known as
the iterative GMM estimator and is obtained with the igmm option. Asymptotically, the two-step and
iterative GMM estimators have the same distribution. However, Hall (2005, 90) suggests that the
iterative estimator may have better finite-sample properties.

Instead of computing Ŵ1 as in (4), we could simply choose Ŵ1 = I, the identity matrix.
The initial estimate, β̂1, would still be consistent. You can request this behavior by specifying
the winitial(identity) option. However, if you specify all your moment equations of the form
E(zu) = 0, we recommend using the default winitial(unadjusted) instead; the rescaling of
the moment conditions implied by using a homoskedastic initial weight matrix makes the numerical
routines used to solve (2) more stable.

If you fit a model with more than one of the moment equations of the form E {h(z;β)} = 0, then
you must use winitial(identity) or winitial(unadjusted, independent). With moment
equations of that form, you do not specify a list of instruments, and gmm cannot evaluate (4)—the
matrix expression in parentheses would necessarily be singular, so it cannot be inverted.

Example 3: Two-step linear GMM estimator

From the previous discussion and the comments in Introduction, we see that the linear 2SLS
estimator is a one-step GMM estimator where we use the weight matrix defined in (4) that assumes
the errors are i.i.d. If we use the 2SLS estimate of β to obtain the sample residuals, compute a new
weight matrix based on those residuals, and then do a second step of GMM estimation, we obtain the
linear two-step GMM estimator as implemented by ivregress gmm.

In example 3 of [R] ivregress, we fit the model of rental rates as discussed in example 2 above.
We now allow the residuals to be heteroskedastic, though we will maintain our assumption that they
are independent. We type

. gmm (rent - {xb:hsngval pcturban} - {b0}), inst(pcturban faminc reg2-reg4)

Step 1
Iteration 0: GMM criterion Q(b) = 56115.03
Iteration 1: GMM criterion Q(b) = 110.91583
Iteration 2: GMM criterion Q(b) = 110.91583

Step 2
Iteration 0: GMM criterion Q(b) = .2406087
Iteration 1: GMM criterion Q(b) = .13672801
Iteration 2: GMM criterion Q(b) = .13672801 (backed up)

GMM estimation

Number of parameters = 3
Number of moments = 6
Initial weight matrix: Unadjusted Number of obs = 50
GMM weight matrix: Robust

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/xb_hsngval .0014643 .0004473 3.27 0.001 .0005877 .002341
/xb_pcturban .7615482 .2895105 2.63 0.009 .1941181 1.328978

/b0 112.1227 10.80234 10.38 0.000 90.95052 133.2949

Instruments for equation 1: pcturban faminc reg2 reg3 reg4 _cons

By default, gmm computes a heteroskedasticity-robust weight matrix before the second step of
estimation, though we could have specified wmatrix(robust) if we wanted to be explicit. Because
we did not specify the vce() option, gmm used a heteroskedasticity-robust one. Our results match
those in example 3 of [R] ivregress. Moreover, the only substantive difference between this example
and example 2 is that here we did not specify the onestep option, so we obtain the two-step
estimates.

http://www.stata.com/manuals13/rivregress.pdf#rivregressRemarksandexamplesivregress_gmmest
http://www.stata.com/manuals13/rivregress.pdf#rivregress
http://www.stata.com/manuals13/rivregress.pdf#rivregressRemarksandexamplesivregress_gmmest
http://www.stata.com/manuals13/rivregress.pdf#rivregress

Obtaining standard errors

This section is a bit more theoretical and can be skipped on first reading. However, the information
is sufficiently important that you should return to this section at some point.

So far in our discussion, we have focused on point estimation without much mention of how we
obtain the standard errors of the estimates. We also mentioned that if we choose W to be the inverse
of the covariance matrix of the moment conditions, then we obtain the “optimal” GMM estimator. We
elaborate those points now.

Using mostly standard statistical arguments, we can show that for the GMM estimator defined in
(2), the variance of β̂ is given by

Var(β̂) =
1

N

{
G(β̂)′WG(β̂)

}−1
G(β̂)′WSWG(β̂)

{
G(β̂)′WG(β̂)

}−1
(5)

where

G(β̂) =
1

N

∑
i
zi
∂ui
∂β

∣∣∣∣
β=β̂

or G(β̂) =
1

N

∑
i

∂hi
∂β

∣∣∣∣
β=β̂

as the case may be and S = E(zuu′z′).

Assuming the vce(unadjusted) option is not specified, gmm reports standard errors based on the
robust variance matrix defined in (5). For the two-step estimator, W is the weight matrix requested
using the wmatrix() option, and it is calculated based on the residuals obtained after the first
estimation step. The second-step point estimates and residuals are obtained, and S is calculated based
on the specification of the vce() option. For the iterated estimator, W is calculated based on the
second-to-last round of estimation, while S is based on the residuals obtained after the last round of
estimation. Computation of the covariance matrix for the one-step estimator is, perhaps surprisingly,
more involved; we discuss the covariance matrix with the one-step estimator in the technical note at
the end of this section.

If we choose the weight matrix to be the inverse of the covariance matrix of the moment conditions
so that W = S−1, then (5) simplifies substantially:

Var(β̂) =
1

N

{
G(β̂)′WG(β̂)

}−1
(6)

The GMM estimator constructed using this choice of weight matrix along with the covariance matrix
in (6) is known as the “optimal” GMM estimator. One can show that if in fact W = S−1, then the
variance in (6) is smaller than the variance in (5) of any other GMM estimator based on the same
moment conditions but with a different choice of weight matrix. Thus the optimal GMM estimator
is also known as the efficient GMM estimator, because it has the smallest variance of any estimator
based on the given moment conditions.

To obtain standard errors from gmm based on the optimal GMM estimator, you specify the
vce(unadjusted) option. We call that VCE unadjusted because we do not recompute the residuals
after estimation to obtain the matrix S required in (5) or allow for the fact that those residuals may
not be i.i.d. Some statistical packages by default report standard errors based on (6) and offer standard
errors based on (5) only as an option or not at all. While the optimal GMM estimator is theoretically
appealing, Cameron and Trivedi (2005, 177) suggest that in finite samples it need not perform better
than the GMM estimator that uses (5) to obtain standard errors.

Technical note

Computing the covariance matrix of the parameters after using the one-step estimator is actually a
bit more complex than after using the two-step or iterative estimators. We can illustrate most of the
intricacies by using linear regression with moment conditions of the form E{x(y − x′β)} = 0.

If you specify winitial(unadjusted) and vce(unadjusted), then the initial weight matrix
will be computed as

Ŵ1 =

(
1

N

∑
i

xix
′
i

)−1
(7)

Moreover, for linear regression, we can show that

G(β̂) =
1

N

∑
i

xix
′
i

so that (6) becomes

Var(β̂) =
1

N


(

1

N

∑
i

xix
′
i

)(
1

N

∑
i

xix
′
i

)−1(
1

N

∑
i

xix
′
i

)
−1

=

(∑
i

xix
′
i

)−1
= (X′X)−1 (8)

However, we know that the nonrobust covariance matrix for the OLS estimator is actually σ̂2(X′X)−1.
What is missing from (8) is the scalar σ̂2, the estimated variance of the residuals. When you use the
one-step estimator and specify winitial(unadjusted), the weight matrix (7) does not include the
σ̂2 term because gmm does not have a consistent estimate of β from which it can then estimate σ2.
The point estimates are still correct, because multiplying the weight matrix by a scalar factor does
not affect the solution to the minimization problem.

To circumvent this issue, if you specify winitial(unadjusted) and vce(unadjusted), gmm
uses the estimated β̂ (which is consistent) to obtain a new unadjusted weight matrix that does include
the term σ̂2 so that evaluating (6) will yield correct standard errors.

If you use the two-step or iterated GMM estimators, this extra effort is not needed to obtain standard
errors because the first-step (and subsequent steps’) estimate of β is consistent and can be used to
estimate σ2 or some other weight matrix based on the wmatrix() option. Straightforward algebra
shows that this extra effort is also not needed if you request any type of adjusted (robust) covariance
matrix with the one-step estimator.

A similar issue arises when you specify winitial(identity) and vce(unadjusted) with the
one-step estimator. Again the solution is to compute an unadjusted weight matrix after obtaining β̂
so that (6) provides the correct standard errors.

We have illustrated the problem and solution using a single-equation linear model. However, the
problem arises whenever you use the one-step estimator with an unadjusted VCE, regardless of the
number of equations; and gmm handles all the details automatically. Computation of Hansen’s J
statistic presents an identical issue, and gmm takes care of that as well.

If you supply your own initial weight matrix by using winitial(matname), then the standard
errors (as well as the J statistic reported by estat overid) are based on that weight matrix. You
should verify that the weight matrix you provide will yield appropriate statistics.

Exponential (Poisson) regression models

Exponential regression models are frequently encountered in applied work. For example, they can
be used as alternatives to linear regression models on log-transformed dependent variables, obviating
the need for post-hoc transformations to obtain predicted values in the original metric of the dependent
variable. When the dependent variable represents a discrete count variable, they are also known as
Poisson regression models; see Cameron and Trivedi (2013).

For now, we consider models of the form

y = exp(x′β) + u (9)

where u is a zero-mean additive error term so that E(y) = exp(x′β). Because the error term is
additive, if x represents strictly exogenous regressors, then we have the population moment condition

E[x{y − exp(x′β)}] = 0 (10)

Moreover, because the number of parameters in the model is equal to the number of instruments,
there is no point to using the two-step GMM estimator.

Example 4: Exponential regression

Cameron and Trivedi (2010, 323) fit a model of the number of doctor visits based on whether the
patient has private insurance, whether the patient has a chronic disease, gender, and income. Here
we fit that model by using gmm. To allow for potential excess dispersion, we will obtain a robust VCE
matrix, which is the default for gmm anyway. We type

. use http://www.stata-press.com/data/r13/docvisits

. gmm (docvis - exp({xb:private chronic female income}+{b0})),
> instruments(private chronic female income) onestep

Step 1
Iteration 0: GMM criterion Q(b) = 16.853973
Iteration 1: GMM criterion Q(b) = 2.2706472
Iteration 2: GMM criterion Q(b) = .19088097
Iteration 3: GMM criterion Q(b) = .00041101
Iteration 4: GMM criterion Q(b) = 3.939e-09
Iteration 5: GMM criterion Q(b) = 6.572e-19

GMM estimation

Number of parameters = 5
Number of moments = 5
Initial weight matrix: Unadjusted Number of obs = 4412

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/xb_private .7986654 .1089891 7.33 0.000 .5850507 1.01228
/xb_chronic 1.091865 .0559888 19.50 0.000 .9821291 1.201601
/xb_female .4925481 .0585298 8.42 0.000 .3778317 .6072644
/xb_income .003557 .0010824 3.29 0.001 .0014356 .0056784

/b0 -.2297263 .1108607 -2.07 0.038 -.4470093 -.0124434

Instruments for equation 1: private chronic female income _cons

Our point estimates agree with those reported by Cameron and Trivedi to at least six significant digits;
the small discrepancies are attributable to different optimization techniques and convergence criteria
being used by gmm and poisson. The standard errors differ by a factor of sqrt(4412/4411) because
gmm uses N in the denominator of the formula for the robust covariance matrix, while the robust
covariance matrix estimator used by poisson uses N − 1.

Technical note
That the GMM and maximum likelihood estimators of the exponential regression model coincide is

not a general property of these two classes of estimators. The maximum likelihood estimator solves
the score equations

1

N

N∑
i=1

∂ ln `i
∂β

= 0

where li is the likelihood for the ith observation. These score equations can be viewed as the sample
analogues of the population moment conditions

E

{
∂ ln `i
∂β

}
= 0

establishing that maximum likelihood estimators represent a subset of the class of GMM estimators.

For the Poisson model,
ln `i = −exp(x′iβ) + yix

′
iβ− ln yi!

so the score equations are

1

N

N∑
i=1

xi {yi − exp(x′iβ)} = 0

which are just the sample moment conditions implied by (10) that we used in the previous example.
That is why our results using gmm match Cameron and Trivedi’s results using poisson.

On the other hand, an intuitive set of moment conditions to consider for GMM estimation of a
probit model is

E[x{y − Φ(x′β)}] = 0

where Φ() is the standard normal cumulative distribution function. Differentiating the likelihood
function for the maximum likelihood probit estimator, we can show that the corresponding score
equations are

1

N

N∑
i=1

[
xi

{
yi
φ(x′iβ)

Φ(x′iβ)
− (1− yi)

φ(x′iβ)

1− Φ(x′iβ)

}]
= 0

where φ() is the standard normal density function. These two moment conditions are not equivalent,
so the maximum likelihood and GMM probit estimators are distinct.

Example 5: Comparison of GMM and maximum likelihood

Using the automobile dataset, here we fit a probit model of foreign on gear ratio, length,
and headroom using first the score equations and then the intuitive set of GMM equations. We type

. use http://www.stata-press.com/data/r13/auto
(1978 Automobile Data)

. global xb "{b1}*gear_ratio + {b2}*length + {b3}*headroom + {b0}"

. global phi "normalden($xb)"

. global Phi "normal($xb)"

. gmm (foreign*$phi/$Phi - (1-foreign)*$phi/(1-$Phi)),
> instruments(gear_ratio length headroom) onestep

(output omitted )
. estimates store ml

. gmm (foreign - $Phi), instruments(gear_ratio length headroom) onestep
(output omitted )

. estimates store gmm

. estimates table ml gmm, b se

Variable ml gmm

b1
_cons 2.9586277 2.8489213

.64042341 .63570247

b2
_cons -.02148933 -.02056033

.01382043 .01396954

b3
_cons .01136927 .02240761

.27278528 .2849891

b0
_cons -6.0222289 -5.8595615

3.5594588 3.5188028

legend: b/se

The coefficients on gear ratio and length are close for the two estimators. The GMM estimate of
the coefficient on headroom is twice that of the maximum likelihood estimate, though the relatively
large standard errors imply that this difference is not significant. You can verify that the coefficients
in the column marked “ml” match those you would obtain using probit. We have not discussed the
differences among standard errors based on the various GMM and maximum-likelihood covariance
matrix estimators to avoid tedious algebra, though you can verify that the robust covariance matrix
after one-step GMM estimation differs by only a finite-sample adjustment factor of (N/N − 1) from
the robust covariance matrix reported by probit. Both the maximum likelihood and GMM probit
estimators require the normality assumption, and the maximum likelihood estimator is efficient if that
normality assumption is correct; therefore, in this particular example, there is no reason to prefer the
GMM estimator.

We can modify (10) easily to allow for endogenous regressors. Suppose that xj is endogenous in
the sense that E(u|xj) 6= 0. Then (10) is no longer a valid moment condition. However, suppose we
have some variables other than x such that E(u|z) = 0. We can instead use the moment conditions

E(zu) = E[z{y − exp(x′β)}] = 0 (11)

As usual, if some elements of x are exogenous, then they should appear in z as well.

Example 6: Exponential regression with endogenous regressors

Returning to the model discussed in example 4, here we treat income as endogenous; unobservable
factors that determine a person’s income may also affect the number of times a person visits a doctor.
We use a person’s age and race as instruments. These are valid instruments if we believe that age
and race influence a person’s income but do not have a direct impact on the number of doctor visits.
(Whether this belief is justified is another matter; we test that belief in [R] gmm postestimation.)
Because we have more instruments (seven) than parameters (five), we have an overidentified model.
Therefore, the choice of weight matrix does matter. We will utilize the default two-step GMM estimator.
In the first step, we will use a weight matrix that assumes the errors are i.i.d. In the second step, we
will use a weight matrix that assumes heteroskedasticity. When you specify twostep, these are the
defaults for the first- and second-step weight matrices, so we do not have to use the winitial() or
wmatrix() options. We will again obtain a robust VCE, which is also the default. We type

. use http://www.stata-press.com/data/r13/docvisits

. gmm (docvis - exp({xb:private chronic female income}+{b0})),
> instruments(private chronic female age black hispanic) twostep

Step 1
Iteration 0: GMM criterion Q(b) = 16.910173
Iteration 1: GMM criterion Q(b) = .82276104
Iteration 2: GMM criterion Q(b) = .21832032
Iteration 3: GMM criterion Q(b) = .12685935
Iteration 4: GMM criterion Q(b) = .12672369
Iteration 5: GMM criterion Q(b) = .12672365

Step 2
Iteration 0: GMM criterion Q(b) = .00234641
Iteration 1: GMM criterion Q(b) = .00215957
Iteration 2: GMM criterion Q(b) = .00215911
Iteration 3: GMM criterion Q(b) = .00215911

GMM estimation

Number of parameters = 5
Number of moments = 7
Initial weight matrix: Unadjusted Number of obs = 4412
GMM weight matrix: Robust

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/xb_private .535335 .1599039 3.35 0.001 .2219291 .8487409
/xb_chronic 1.090126 .0617659 17.65 0.000 .9690668 1.211185
/xb_female .6636579 .0959884 6.91 0.000 .4755241 .8517918
/xb_income .0142855 .0027162 5.26 0.000 .0089618 .0196092

/b0 -.5983477 .138433 -4.32 0.000 -.8696713 -.327024

Instruments for equation 1: private chronic female age black hispanic _cons

Once we control for the endogeneity of income, we find that its coefficient has quadrupled in size.
Additionally, access to private insurance has less of an impact on the number of doctor visits and
gender has more of an impact.

http://www.stata.com/manuals13/rgmmpostestimation.pdf#rgmmpostestimation


24 gmm — Generalized method of moments estimation

Technical note
Although perhaps at first tempting, unlike the Poisson model, you cannot simply replace x in

the moment conditions for the probit (or logit) model with a vector of instruments, z, if you have
endogenous regressors. See Wilde (2008).

Mullahy (1997) considers a slightly more complicated version of the exponential regression model
that incorporates nonadditive unobserved heterogeneity. His model can be written as

yi = exp(x′iβ)ηi + εi

where ηi > 0 is an unobserved heterogeneity term that may be correlated with xi. One result from
his paper is that instead of using the additive moment condition (10), we can use the multiplicative
moment condition

E

{
z
y − exp(x′β)

exp(x′β)

}
= E[z{yexp(−x′β)− 1}] = 0 (12)

Windmeijer and Santos Silva (1997) discuss the use of additive versus multiplicative moment conditions
with endogenous regressors and note that a set of instruments that satisfies the additive moment
conditions will not also satisfy the multiplicative moment conditions. They remark that which to
use is an empirical issue that can at least partially be settled by using the test of overidentifying
restrictions that is implemented by estat overid after gmm to ascertain whether the instruments for
a given model are valid. See [R] gmm postestimation for information on the test of overidentifying
restrictions.

Specifying derivatives

By default, gmm calculates derivatives numerically, and the method used produces accurate results
for the vast majority of applications. However, if you refit the same model repeatedly or else have
the derivatives available, then gmm will run more quickly if you supply it with analytic derivatives.

When you use the interactive version of gmm, you specify derivatives using substitutable expressions
in much the same way you specify the moment equations. There are three rules you must follow:

1. As with the substitutable expressions that define residual equations, you bind parameters of
the model in braces: {b0}, {param}, etc.

2. You must specify a derivative for each parameter that appears in each moment equation. If
a parameter does not appear in a moment equation, then you do not specify a derivative for
that parameter in that moment equation.

3. If you declare a linear combination in an equation, then you specify a derivative with respect
to that linear combination. gmm applies the chain rule to obtain the derivatives with respect
to the individual parameters encompassed by that linear combination.

We illustrate with several examples.

Example 7: Derivatives for a single-equation model

Consider a simple exponential regression model with one exogenous regressor and a constant term.
We have

ui = yi − exp(β0 + β1xi)

Now
∂ui
∂β0

= −exp(β0 + β1xi) and
∂ui
∂β1

= −xi exp(β0 + β1xi)

http://www.stata.com/manuals13/rgmmpostestimation.pdf#rgmmpostestimation

In Stata, we type

. gmm (docvis - exp({b0} + {b1}*income)), instruments(income)
> deriv(/b0 = -1*exp({b0} + {b1}*income))
> deriv(/b1 = -1*income*exp({b0}+{b1}*income)) onestep

Step 1
Iteration 0: GMM criterion Q(b) = 9.1548611
Iteration 1: GMM criterion Q(b) = 3.5146131
Iteration 2: GMM criterion Q(b) = .01344695
Iteration 3: GMM criterion Q(b) = 3.690e-06
Iteration 4: GMM criterion Q(b) = 4.606e-13
Iteration 5: GMM criterion Q(b) = 1.501e-26

GMM estimation

Number of parameters = 2
Number of moments = 2
Initial weight matrix: Unadjusted Number of obs = 4412

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/b0 1.204888 .0462355 26.06 0.000 1.114268 1.295507
/b1 .0046702 .0009715 4.81 0.000 .0027662 .0065743

Instruments for equation 1: income _cons

Notice how we specified the derivative() option for each parameter. We simply specified a slash,
the name of the parameter, an equal sign, then a substitutable expression that represents the derivative.
Because our model has only one residual equation, we do not need to specify equation numbers in
the derivative() options.

When you specify a linear combination of variables, your derivative should be with respect to the
entire linear combination. For example, say we have the residual equation

ui = y − exp(x′iβ + β0)

for which we would type

. gmm (y - exp({xb: x1 x2 x3} + {b0}) ...

Then in addition to the derivative ∂ui/∂β0, we are to compute and specify

∂ui
∂x′iβ

= −exp(x′iβ + β0)

Using the chain rule, ∂ui/∂βj = ∂ui/∂(x′iβ)×∂(x′iβ)/∂βj = −xijexp(x′iβ+β0). Stata does this
last calculation automatically. It knows the variables in the linear combination, so all it needs is the
derivative of the residual function with respect to the linear combination. This allows you to change
the variables in your linear combination without having to change the derivatives.

Example 8: Derivatives with a linear combination

We refit the model described in the example illustrating exponential regression with endogenous
regressors, now providing analytic derivatives. We type

. gmm (docvis - exp({xb:private chronic female income}+{b0})),
> instruments(private chronic female age black hispanic)
> derivative(/xb = -1*exp({xb:} + {b0}))
> derivative(/b0 = -1*exp({xb:} + {b0}))

Step 1
Iteration 0: GMM criterion Q(b) = 16.910173
Iteration 1: GMM criterion Q(b) = .82270871
Iteration 2: GMM criterion Q(b) = .21831995
Iteration 3: GMM criterion Q(b) = .12685934
Iteration 4: GMM criterion Q(b) = .12672369
Iteration 5: GMM criterion Q(b) = .12672365

Step 2
Iteration 0: GMM criterion Q(b) = .00234641
Iteration 1: GMM criterion Q(b) = .00215957
Iteration 2: GMM criterion Q(b) = .00215911
Iteration 3: GMM criterion Q(b) = .00215911

GMM estimation

Number of parameters = 5
Number of moments = 7
Initial weight matrix: Unadjusted Number of obs = 4412
GMM weight matrix: Robust

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/xb_private .535335 .159904 3.35 0.001 .221929 .848741
/xb_chronic 1.090126 .0617659 17.65 0.000 .9690668 1.211185
/xb_female .6636579 .0959885 6.91 0.000 .475524 .8517918
/xb_income .0142855 .0027162 5.26 0.000 .0089618 .0196092

/b0 -.5983477 .138433 -4.32 0.000 -.8696714 -.327024

Instruments for equation 1: private chronic female age black hispanic _cons

In the first derivative() option, we specified the name of the linear combination, xb, instead of
an individual parameter’s name. We already declared the variables of our linear combination in the
substitutable expression for the residual equation, so in our substitutable expressions for the derivatives,
we can use the shorthand notation {xb:} to refer to it.

Our point estimates are identical to those we obtained earlier. The standard errors and confidence
intervals differ by only trivial amounts.

Exponential regression models with panel data
In addition to supporting cross-sectional and time-series data, gmm also works with panel-data

models. Here we illustrate gmm’s panel-data capabilities by expanding our discussion of exponential
regression models to allow for panel data. This also provides us the opportunity to demonstrate
the moment-evaluator program version of gmm. Our discussion is based on Blundell, Griffith, and
Windmeijer (2002). Also see Wooldridge (1999) for further discussion of nonlinear panel-data models.

First, we expand (9) for panel data. With individual heterogeneity term ηi, we have

E(yit|xit, ηi) = exp(x′itβ + ηi) = µitνi

where µit = exp(x′itβ) and νi = exp(ηi). Note that there is no constant term in this model because
its effect cannot be disentangled from νi. With an additive idiosyncratic error term, we have the
regression model

yit = µitνi + εit

We do not impose the assumption E(xitηi) = 0, so ηi can be considered a fixed effect in the sense
that it may be correlated with the regressors.

As discussed by Blundell, Griffith, and Windmeijer (2002), if xit is strictly exogenous, meaning
E(xitεis) = 0 for all t and s, then we can estimate the parameters of the model by using the sample
moment conditions ∑

i

∑
t

xit

(
yit − µit

yi
µi

)
= 0 (13)

where yi and µi are the means of yit and µit for panel i, respectively. Because µi depends on
the parameters of the model, it must be recomputed each time gmm needs to evaluate the residual
equation. Therefore, we cannot use the substitutable expression version of gmm. Instead, we must use
the moment-evaluator program version.

The moment-evaluator program version of gmm functions much like the function-evaluator program
versions of nl and nlsur. The program you write is passed one or more variables to be filled in with
the residuals evaluated at the parameter values specified in an option passed to your program. For the
fixed-effects Poisson model with strictly exogenous regressors, our first crack at a function-evaluator
program is

program gmm_poi

version 13
syntax varlist if, at(name)

quietly {
tempvar mu mubar ybar
gen double ‘mu’ = exp(x1*‘at’[1,1] + x2*‘at’[1,2] ///

+ x3*‘at’[1,3]) ‘if’
egen double ‘mubar’ = mean(‘mu’) ‘if’, by(id)
egen double ‘ybar’ = mean(y) ‘if’, by(id)
replace ‘varlist’ = y - ‘mu’*‘ybar’/‘mubar’ ‘if’

}

end

You can save your program in an ado-file named name.ado, where name is the name you use for
your program; here we would save the program in the ado-file gmm poi.ado. Alternatively, if you
are working from within a do-file, you can simply define the program before calling gmm. The syntax
statement declares that we are expecting to receive a varlist, containing the names of variables whose
values we are to replace with the values of the residual equations, and an if expression that will
mark the estimation sample; because our model has one residual equation, varlist will consist of one
variable. at() is a required option to our program, and it will contain the name of a matrix containing
the parameter values at which we are to evaluate the residual equation. All moment-evaluator programs
must accept the varlist, if condition, and at() option.

The first part of our program computes µit. In the model we will fit shortly, we have three
regressors, named x1, x2, and x3. The ‘at’ vector will have three elements, one for each of those
variables. Notice that we included ‘if’ at the end of each statement that affects variables to restrict
the computations to the relevant estimation sample. The two egen statements compute µi and yi;
in the example dataset we will use shortly, the panel variable is named id, and for simplicity we
hardcoded that variable into our program as well. Finally, we compute the residual equation, which
is the portion of (13) bound in parentheses.

Example 9: Panel Poisson with strictly exogenous regressors

To fit our model, we type

. use http://www.stata-press.com/data/r13/poisson1

. gmm gmm_poi, nequations(1) parameters(b1 b2 b3)
> instruments(x1 x2 x3, noconstant) vce(cluster id) onestep

Step 1
Iteration 0: GMM criterion Q(b) = 51.99142
Iteration 1: GMM criterion Q(b) = .04345191
Iteration 2: GMM criterion Q(b) = 8.720e-06
Iteration 3: GMM criterion Q(b) = 7.115e-13
Iteration 4: GMM criterion Q(b) = 5.130e-27

GMM estimation

Number of parameters = 3
Number of moments = 3
Initial weight matrix: Unadjusted Number of obs = 409

(Std. Err. adjusted for 45 clusters in id)

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/b1 1.94866 .1000265 19.48 0.000 1.752612 2.144709
/b2 -2.966119 .0923592 -32.12 0.000 -3.14714 -2.785099
/b3 1.008634 .1156561 8.72 0.000 .781952 1.235315

Instruments for equation 1: x1 x2 x3

All three of our regressors are strictly exogenous, so they can serve as their own regressors.
There is no constant term in the model (it would be unidentified), so we exclude a constant term
from our list of instruments. We have one residual equation as indicated by nequations(1), and
we have three parameters, named b1, b2, and b3. The order in which you declare parameters in
the parameters() option determines the order in which they appear in the ‘at’ vector in the
moment-evaluator program. We specified vce(cluster id) to obtain standard errors that allow for
correlation among observations within each panel.

The program we just wrote is sufficient to fit the model to the poisson1 dataset, but if we want
to fit that model to other datasets, we would need to change the variable names and perhaps account
for having a different number of parameters as well. Despite those limitations, if you just want to fit
a single model, that program is adequate.

Next we take advantage of the ability to specify full equation names in the parameters() option
and rewrite our evaluator program so that we can more easily change the variables in our model.
This approach is particularly useful if some of the moment equations are linear in the parameters,
because then we can use matrix score (see [P] matrix score) to evaluate those moments.

http://www.stata.com/manuals13/pmatrixscore.pdf#pmatrixscore

Our new evaluator program is

program gmm_poieq

version 13
syntax varlist if, at(name)

quietly {
tempvar mu mubar ybar

matrix score double ‘mu’ = ‘at’ ‘if’, eq(#1)
replace ‘mu’ = exp(‘mu’)

egen double ‘mubar’ = mean(‘mu’) ‘if’, by(id)
egen double ‘ybar’ = mean(y) ‘if’, by(id)
replace ‘varlist’ = y - ‘mu’*‘ybar’/‘mubar’ ‘if’

}

end

Rather than using generate to compute the temporary variable ‘mu’, we used matrix score to
obtain the linear combination x′itβ and then called replace to compute exp(x′itβ).

Example 10: Panel Poisson using matrix score

To fit our model, we type

. use http://www.stata-press.com/data/r13/poisson1

. gmm gmm_poieq, nequations(1) parameters(y:x1 y:x2 y:x3)
> instruments(x1 x2 x3, noconstant) vce(cluster id) onestep

Step 1
Iteration 0: GMM criterion Q(b) = 51.99142
Iteration 1: GMM criterion Q(b) = .04345191
Iteration 2: GMM criterion Q(b) = 8.720e-06
Iteration 3: GMM criterion Q(b) = 7.115e-13
Iteration 4: GMM criterion Q(b) = 5.106e-27

GMM estimation

Number of parameters = 3
Number of moments = 3
Initial weight matrix: Unadjusted Number of obs = 409

(Std. Err. adjusted for 45 clusters in id)

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

x1 1.94866 .1000265 19.48 0.000 1.752612 2.144709
x2 -2.966119 .0923592 -32.12 0.000 -3.14714 -2.785099
x3 1.008634 .1156561 8.72 0.000 .781952 1.235315

Instruments for equation 1: x1 x2 x3

Instead of specifying simple parameter names in the parameters() option, we specified an
equation name and the variables associated with that equation. We named our equation y, but you
could use any valid Stata name. When we use this syntax, the rows of the coefficient table are grouped
by the equation names.

Say we wanted to refit our model using just x1 and x3 as regressors. We do not need to make
any changes to gmm poieq. We just change the specification of the parameters() option:

. gmm gmm_poieq, nequations(1) parameters(y:x1 y:x3) ///
> instruments(x1 x3, noconstant) vce(cluster id) onestep

In this evaluator program, we have still hard-coded the name of the dependent variable. The next two
examples include methods to tackle that shortcoming.

Technical note
Say we specify the parameters() option like this:

. gmm . . ., parameters(y1:x1 y1:x2 y1:_cons y2:_cons y3:x1 y3:_cons)

Then the ‘at’ vector passed to our program will have the following column names attached to it:

‘at’[1,6]
y1: y1: y1: y2: y3: y3:
x1 x2 _cons _cons x1 _cons

Typing

. matrix score double eq1 = ‘at’, eq(#1)

is equivalent to typing

. generate double eq1 = x1*‘at’[1,1] + x2*‘at’[1,2] + ‘at’[1,3]

with one important difference. If we change some of the variables in the parameters() option when
we call gmm, matrix score will compute the correct linear combination. If we were to use the
generate statement instead, then every time we wanted to change the variables in our model, we
would have to modify that statement as well.

The command

. matrix score double alpha = ‘at’, eq(#2) scalar

is equivalent to

. scalar alpha = ‘at’[1,4]

Thus even if you specify equation and variable names in the parameters() option, you can still
have scalar parameters in your model.

When past values of the idiosyncratic error term affect the value of a regressor, we say that regressor
is predetermined. When one or more regressors are predetermined, sample moment condition (10) is
no longer valid. However, Chamberlain (1992) shows that a simple alternative is to consider moment
conditions of the form ∑

i

T∑
t=2

xi,t−1

(
yi,t−1 − µi,t−1

yit
µit

)
= 0 (14)

Also see Wooldridge (1997) and Windmeijer (2000) for other moment conditions that can be used
with predetermined regressors.

Example 11: Panel Poisson with predetermined regressors

Here we refit the previous model, treating all the regressors as predetermined and using the moment
conditions in (14). Our moment-evaluator program is

program gmm_poipre

version 13
syntax varlist if, at(name) mylhs(varlist)

quietly {
tempvar mu

matrix score double ‘mu’ = ‘at’ ‘if’, eq(#1)
replace ‘mu’ = exp(‘mu’)

replace ‘varlist’ = L.‘mylhs’ - L.‘mu’*‘mylhs’/‘mu’ ‘if’
}

end

To compute the moment equation, we used lag-operator notation so that Stata properly handles gaps
in our panel dataset. We also made our program accept an additional option that we will use to pass
in the dependent variable. When we specify this option in our gmm statement, it will get passed to
our evaluator program because gmm will not recognize the option as one of its own. Equation (14)
shows that we are to use the first lags of the regressors as instruments, so we type

. gmm gmm_poipre, mylhs(y) nequations(1) vce(cluster id) onestep
> parameters(y:x1 y:x2 y:x3) instruments(L.(x1 x2 x3), noconstant)
warning: 45 missing values returned for equation 1 at initial values

Step 1
Iteration 0: GMM criterion Q(b) = 52.288048
Iteration 1: GMM criterion Q(b) = 2.3599714
Iteration 2: GMM criterion Q(b) = .16951739
Iteration 3: GMM criterion Q(b) = .00020399
Iteration 4: GMM criterion Q(b) = 3.392e-10
Iteration 5: GMM criterion Q(b) = 9.230e-22

GMM estimation

Number of parameters = 3
Number of moments = 3
Initial weight matrix: Unadjusted Number of obs = 319

(Std. Err. adjusted for 45 clusters in id)

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

x1 2.025956 .2777156 7.30 0.000 1.481644 2.570269
x2 -2.909646 .2577577 -11.29 0.000 -3.414842 -2.404451
x3 1.202926 .1873571 6.42 0.000 .8357131 1.570139

Instruments for equation 1: L.x1 L.x2 L.x3

Here, like earlier with strictly exogenous regressors, the number of instruments equals the number of
parameters, so there is no gain to using the two-step or iterated estimators. However, if you do have
more instruments than parameters, you will most likely want to use one of those other estimators
instead.

Instead of making our program accept the mylhs() option, we could have used Stata’s coleq
extended macro function to determine the dependent variable based on the column names attached to
the ‘at’ vector; see [P] macro. Then we could refit our model with a different dependent variable by
changing the eqname used in the parameters() option. In the next example, we take this approach.

http://www.stata.com/manuals13/pmacro.pdf#pmacro

In the previous example, we used xi,t−1 as instruments. A more efficient GMM estimator would
also use xi,t−2,xi,t−3, . . . ,xi,1 as instruments in period t as well. gmm’s xtinstruments() option
allows you to specify instrument lists that grow as t increases. Later we discuss the xtinstruments()
option in detail in the context of linear dynamic panel-data models.

When a regressor is contemporaneously correlated with the idiosyncratic error term, we say that
regressor is endogenous. Windmeijer (2000) shows that here we can use the moment condition

∑
i

T∑
t=3

xi,t−2

(
yit
µit
− yi,t−1
µi,t−1

)
Here we use the second lag of the endogenous regressor as an instrument. If a variable is strictly
exogenous, it can of course serve as its own instrument.

Example 12: Panel Poisson with endogenous regressors

Here we refit the model, treating x3 as endogenous and x1 and x2 as strictly exogenous. Our
moment-evaluator program is

program gmm_poiend

version 13
syntax varlist if, at(name)

quietly {

tempvar mu

matrix score double ‘mu’ = ‘at’ ‘if’, eq(#1)
replace ‘mu’ = exp(‘mu’)
local mylhs : coleq ‘at’
local mylhs : word 1 of ‘mylhs’

replace ‘varlist’ = ‘mylhs’/‘mu’ - L.‘mylhs’/L.‘mu’ ‘if’
}

end

Now we call gmm using x1, x2, and L2.x3 as instruments:

. use http://www.stata-press.com/data/r13/poisson2

. gmm gmm_poiend, nequations(1) vce(cluster id) onestep
> parameters(y:x1 y:x2 y:x3) instruments(x1 x2 L2.x3, noconstant)
warning: 500 missing values returned for equation 1 at initial values

Step 1
Iteration 0: GMM criterion Q(b) = 43.799922
Iteration 1: GMM criterion Q(b) = .06998898
Iteration 2: GMM criterion Q(b) = .04165161
Iteration 3: GMM criterion Q(b) = .03573502
Iteration 4: GMM criterion Q(b) = .00001981
Iteration 5: GMM criterion Q(b) = 3.168e-12
Iteration 6: GMM criterion Q(b) = 1.529e-23

GMM estimation

Number of parameters = 3
Number of moments = 3
Initial weight matrix: Unadjusted Number of obs = 3266

(Std. Err. adjusted for 500 clusters in id)

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

x1 1.857766 .2631454 7.06 0.000 1.34201 2.373521
x2 -2.865858 .2151529 -13.32 0.000 -3.28755 -2.444167
x3 4.961867 14.91462 0.33 0.739 -24.27025 34.19399

Instruments for equation 1: x1 x2 L2.x3

The warning at the top of the output indicates that we have 500 panels in our dataset. Our moment
equation includes lagged terms and therefore cannot be evaluated for the first time period within each
panel. Warning messages like that can be ignored once you know why they occurred. If you receive
a warning message that you were not expecting, you should first investigate the cause of the warning
before trusting the results. As in the previous example, instead of using just xi,t−2 as an instrument,
we could use all further lags of xit as instruments as well.

Rational-expectations models
Macroeconomic models typically assume that agents’ expectations about the future are formed

rationally. By rational expectations, we mean that agents use all information available when forming
their forecasts, so the forecast error is uncorrelated with the information available when the forecast was
made. Say that at time t, people make a forecast, ŷt+1, of variable y in the next period. If Ωt denotes
all available information at time t, then rational expectations implies that E {(ŷt+1 − yt+1)|Ωt} = 0.
If Ωt denotes observable variables such as interest rates or prices, then this conditional expectation
can serve as the basis of a moment condition for GMM estimation.

Example 13: Fitting a Euler equation

In a well-known article, Hansen and Singleton (1982) consider a model of portfolio decision
making and discuss parameter estimation using GMM. We will consider a simple example with one
asset in which the agent can invest. A consumer wants to maximize the present value of his lifetime
utility derived from consuming the good. On the one hand, the consumer is impatient, so he would
rather consume today than wait until tomorrow. On the other hand, if he consumes less today, he can
invest more of his money, earning more interest that he can then use to consume more of the good
tomorrow. Thus there is a tradeoff between having his cake today or sacrificing a bit today to have
more cake tomorrow.

If we assume a specific form for the agent’s utility function, known as the constant relative-risk
aversion utility function, we can show that the Euler equation is

E
[
zt
{

1− β(1 + rt+1)(ct+1/ct)
−γ}] = 0

where β and γ are the parameters to estimate, rt is the return to the financial asset, and ct is
consumption in period t. β measures the agent’s discount factor. If β is near one, the agent is patient
and is more willing to forgo consumption this period. If β is close to zero, the agent is less patient
and prefers to consume more now. The parameter γ characterizes the agent’s utility function. If γ
equals zero, the utility function is linear. As γ tends toward one, the utility function tends toward
u = log(c).

We have data on 3-month Treasury bills (rt) and consumption expenditures (ct). As instruments,
we will use lagged rates of return and past growth rates of consumption. We will use the two-step
estimator and a weight matrix that allows for heteroskedasticity and autocorrelation up to four lags
with the Bartlett kernel. In Stata, we type

. use http://www.stata-press.com/data/r13/cr

. generate cgrowth = c / L.c
(1 missing value generated)

. gmm (1 - {b=1}*(1+F.r)*(F.c/c)^(-1*{gamma=1})),
> inst(L.r L2.r cgrowth L.cgrowth) wmat(hac nw 4) twostep
warning: 1 missing value returned for equation 1 at initial values

Step 1
Iteration 0: GMM criterion Q(b) = .00226482
Iteration 1: GMM criterion Q(b) = .00054369
Iteration 2: GMM criterion Q(b) = .00053904
Iteration 3: GMM criterion Q(b) = .00053904

Step 2
Iteration 0: GMM criterion Q(b) = .0600729
Iteration 1: GMM criterion Q(b) = .0596369
Iteration 2: GMM criterion Q(b) = .0596369

GMM estimation

Number of parameters = 2
Number of moments = 5
Initial weight matrix: Unadjusted Number of obs = 239
GMM weight matrix: HAC Bartlett 4

HAC
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/b .9204617 .0134646 68.36 0.000 .8940716 .9468518
/gamma -4.222361 1.473895 -2.86 0.004 -7.111143 -1.333579

HAC standard errors based on Bartlett kernel with 4 lags.
Instruments for equation 1: L.r L2.r cgrowth L.cgrowth _cons

The warning message at the top of the output appears because the forward operator in our substitutable
expression says that residuals can be computed only for 239 observations; our dataset contains 240
observations. Our estimate of β is near one, in line with expectations and published results. However,
our estimate of γ implies risk-loving behavior and therefore a poorly specified model.

System estimators

In many economic models, two or more variables are determined jointly through a system of
simultaneous equations. Indeed, some of the earliest work in econometrics, including that of the
Cowles Commission, was centered around estimation of the parameters of simultaneous equations.
The 2SLS and IV estimators we have already discussed are used in some circumstances to estimate
such parameters. Here we focus on the joint estimation of all the parameters of systems of equations,
and we begin with the well-known three-stage least-squares (3SLS) estimator.

Recall that the 2SLS estimator is based on the moment conditions E(zu) = 0. The 2SLS estimator
can be used to estimate the parameters of one equation of a system of structural equations. Moreover,
with the 2SLS estimator, we do not even need to specify the structural relationship among all the
endogenous variables; we need to specify only the equation on which interest focuses and simply
assume reduced-form relationships among the endogenous regressors of the equation of interest and
the exogenous variables of the model. If we are willing to specify the complete system of structural
equations, then assuming our model is correctly specified, by estimating all the equations jointly, we
can obtain estimates that are more efficient than equation-by-equation 2SLS.

In [R] reg3, we fit a simple two-equation macroeconomic model:

consump = β0 + β1wagepriv + β2wagegovt + ε1 (15)

wagepriv = β3 + β4consump + β5govt + β6capital1 + ε2 (16)

where consump represents aggregate consumption; wagepriv and wagegovt are total wages paid
by the private and government sectors, respectively; govt is government spending; and capital1 is
the previous period’s capital stock. We are not willing to assume that ε1 and ε2 are independent, so
we must treat both consump and wagepriv as endogenous. Suppose that a random shock makes ε2
positive. Then by (16), wagepriv will be higher than it otherwise would. Moreover, ε1 will either
be higher or lower, depending on the correlation between it and ε2. The shock to ε2 has made both
wagepriv and ε1 move, implying that in (15) wagepriv is an endogenous regressor. A similar
argument shows that consump is an endogenous regressor in the second equation. In our model,
wagegovt, govt, and capital1 are all exogenous variables.

Let z1 and z2 denote the instruments for the first and second equations, respectively; we will
discuss what comprises them shortly. We have two sets of moment conditions:

E

{
z1(consump− β0 − β1wagepriv− β2wagegovt)

z2(wagepriv− β3 − β4consump− β5govt− β6capital1)

}
= 0 (17)

One of the defining characteristics of 3SLS is that the errors are homoskedastic conditional on the
instrumental variables. Using this assumption, we have

E

[{
z1ε1
z2ε2

}
{ z′1ε1 z′2ε2 }

]
=

{
σ11E(z1z

′
1) σ12E(z1z

′
2)

σ21E(z2z
′
1) σ22E(z2z

′
2)

}
(18)

where σij = cov(εi, εj). Let Σ denote the 2× 2 matrix with typical element σij .

The second defining characteristic of the 3SLS estimator is that it uses all the exogenous variables
as instruments for all equations; here z1 = z2 = (wagegovt, govt, capital1, 1), where the 1
indicates a constant term. From our discussion on the weight matrix and two-step estimation, we
want to use the sample analogue of the matrix inverse of the right-hand side of (18) as our weight
matrix.

To implement the 3SLS estimator, we apparently need to know Σ or at least have a consistent
estimator of it. The solution is to fit (15) and (16) by 2SLS, use the sample residuals ε̂1 and ε̂2 to
estimate Σ, then estimate the parameters of (17) via GMM by using the weight matrix just discussed.

http://www.stata.com/manuals13/rreg3.pdf#rreg3

Example 14: 3SLS estimation

3SLS is easier to do using gmm than it sounds. The 3SLS estimator is a two-step GMM estimator. In
the first step, we do the equivalent of 2SLS on each equation, and then we compute a weight matrix
based on (18). Finally, we perform a second step of GMM with this weight matrix.

In Stata, we type
. use http://www.stata-press.com/data/r13/klein, clear

. gmm (eq1: consump - {b0} - {xb: wagepriv wagegovt})
> (eq2: wagepriv - {c0} - {xc: consump govt capital1}),
> instruments(eq1: wagegovt govt capital1)
> instruments(eq2: wagegovt govt capital1)
> winitial(unadjusted, independent) wmatrix(unadjusted) twostep

Step 1
Iteration 0: GMM criterion Q(b) = 4195.4487
Iteration 1: GMM criterion Q(b) = .22175631
Iteration 2: GMM criterion Q(b) = .22175631 (backed up)

Step 2
Iteration 0: GMM criterion Q(b) = .09716589
Iteration 1: GMM criterion Q(b) = .07028208
Iteration 2: GMM criterion Q(b) = .07028208

GMM estimation

Number of parameters = 7
Number of moments = 8
Initial weight matrix: Unadjusted Number of obs = 22
GMM weight matrix: Unadjusted

Coef. Std. Err. z P>|z| [95% Conf. Interval]

/b0 19.3559 3.583772 5.40 0.000 12.33184 26.37996
/xb_wagepriv .8012754 .1279329 6.26 0.000 .5505314 1.052019
/xb_wagegovt 1.029531 .3048424 3.38 0.001 .432051 1.627011

/c0 14.63026 10.26693 1.42 0.154 -5.492552 34.75306
/xc_consump .4026076 .2567312 1.57 0.117 -.1005764 .9057916

/xc_govt 1.177792 .5421253 2.17 0.030 .1152461 2.240338
/xc_capital1 -.0281145 .0572111 -0.49 0.623 -.1402462 .0840173

Instruments for equation 1: wagegovt govt capital1 _cons
Instruments for equation 2: wagegovt govt capital1 _cons

The independent suboption of the winitial() option tells gmm to assume that the residuals
are independent across equations; this suboption sets σ21 = σ12 = 0 in (18). Assuming both
homoskedasticity and cross-equation independence is equivalent to fitting the two equations of our
model independently by 2SLS. The wmatrix() option controls how the weight matrix is computed
based on the first-step parameter estimates before the second step of estimation; here we request a
weight matrix that assumes conditional homoskedasticity but that does not impose the cross-equation
independence like the initial weight matrix we used. In this example, we also illustrated how to
name equations and how equation names can be used in the instruments() option. Our results are
identical to those in [R] reg3.

We could have specified our instruments with the syntax
instruments(wagegovt govt capital1)

because gmm uses the variables listed in the instruments() option for all equations unless you
specify which equations the list of instruments is to be used with. However, we wanted to emphasize
that the same instruments are being used for both equations; in a moment, we will discuss an estimator
that does not use the same instruments in all equations.

http://www.stata.com/manuals13/rreg3.pdf#rreg3

In the previous example, if we omit the twostep option, the resulting coefficients will be
equivalent to equation-by-equation 2SLS, which Wooldridge (2010, 216) calls the “system 2SLS
estimator”. Eliminating the twostep option makes the wmatrix() option irrelevant, so that option
can be eliminated as well.

So far, we have developed the traditional 3SLS estimator. Wooldridge (2010, chap. 8) discusses the
“GMM 3SLS” estimator that extends the traditional 3SLS estimator by allowing for heteroskedasticity
and different instruments for different equations.

Generalizing (18) to an arbitrary number of equations, we have

E (Z′εε′Z) = E (Z′ΣZ) (19)

where

Z =


z1 0 · · · 0
0 z2 · · · 0
...

...
. . .

...
0 0 · · · zm


and Σ is now m×m. Equation (19) is the multivariate analogue of a homoskedasticity assumption;
for each equation, the error variance is constant for all observations, as is the covariance between
any two equations’ errors.

We can relax this homoskedasticity assumption by considering different weight matrices. For
example, if we continue to assume that observations are independent but not necessarily identically
distributed, then by specifying wmatrix(robust), we would obtain a weight matrix that allows for
heteroskedasticity:

Ŵ =
1

N

∑
i

Z′iε̂iε̂
′
iZi

This is the weight matrix in Wooldridge’s (2010, 218) Procedure 8.1, “GMM with Optimal Weighting
Matrix”. By default, gmm would report standard errors based on his covariance matrix (8.27); specifying
vce(unadjusted) would provide the optimal GMM standard errors. If you have multiple observations
for each individual or firm in your dataset, you could specify wmatrix(cluster id), where id
identifies individuals or firms. This would allow arbitrary within-individual correlation, though it does
not account for an individual-specific fixed or random effect. In both cases, we would continue to
use winitial(unadjusted, independent) so that the first-step estimates are the system 2SLS
estimates.

Wooldridge (2010, sec. 9.6) discusses instances where it is necessary to use different instruments
in different equations. The GMM 3SLS estimator with different instruments in different equations but
with conditional homoskedasticity is what Hayashi (2000, 275) calls the “full-information instrumental
variables efficient” (FIVE) estimator. Implementing the FIVE estimator is easy with gmm. For example,
say we have a two-equation system, where kids, age, income, and education are all valid
instruments for the first equation; but education is not a valid instrument for the second equation.
Then our syntax would take the form

gmm (<mexp 1>) (<mexp 2>), instruments(1:kids age income education)
instruments(2:kids age income)

The following syntax is equivalent:

gmm (<mexp 1>) (<mexp 2>), instruments(kids age income)
instruments(1:education)

Because we did not specify a list of equations in the second example’s first instruments() option,
those variables are used as instruments in both equations. You can use whichever syntax you prefer.
The first requires a bit more typing but is arguably more transparent.

If all the regressors in the model are exogenous, then the traditional 3SLS estimator is the seemingly
unrelated regression (SUR) estimator. Here you would specify all the regressors as instruments.

Dynamic panel-data models

Commands in Stata that work with panel data expect the data to be in the “long” format, meaning
that each row of the dataset consists of one subobservation that is a member of a logical observation
(represented by the panel identifier variable). See [D] reshape for a discussion of the long versus
“wide” data forms. gmm is no exception in this respect when used with panel data. From a theoretical
perspective, however, it is sometimes easier to view GMM estimators for panel data as system estimators
in which we have N observations on a system of T equations, where N and T are the number of
observations and panels, respectively, rather than a single-equation estimator with NT observations.
Usually, each of the T equations will in fact be the same, though we will want to specify different
instruments for each of these equations.

In a dynamic panel-data model, lagged values of the dependent variable are included as regressors.
Here we consider a simple model with one lag of the dependent variable y as a regressor and a vector
of strictly exogenous regressors, xit:

yit = ρyi,t−1 + x′itβ + ui + εit (20)

ui can be either a fixed- or a random-effect term, in the sense that we do not require xit to be
independent of it. Even with the assumption that εit is i.i.d., the presence of both yi,t−1 and ui in
(20) renders both the standard fixed- and random-effects estimators to be inconsistent because of the
well-known Nickell (1981) bias. OLS regression of yit on yi,t−1 and xit also produces inconsistent
estimates, because yi,t−1 will be correlated with the error term.

Technical note
Stata has the xtabond, xtdpd, and xtdpdsys commands (see [XT] xtabond, [XT] xtdpd, and

[XT] xtdpdsys) to fit equations like (20), and for everyday use those commands are preferred because
they offer features such as Windmeijer (2005) bias-corrected standard errors to account for the bias
of traditional two-step GMM standard errors seen in dynamic panel-data models and, being linear
estimators, only require you to specify variable names instead of complete equations. However, using
gmm has several pedagogical advantages, including the ability to tie those model-specific commands
into a more general framework, a clear illustration of how certain types of instrument matrices for
panel-data models are formed, and demonstrations of several advanced features of gmm.

First-differencing (20) removes the panel-specific ui term:

yit − yi,t−1 = ρ(yi,t−1 − yi,t−2) + (xit − xi,t−1)′β + (εit − εi,t−1) (21)

However, now (yi,t−1− yi,t−2) is correlated with (εit− εi,t−1). Thus we need an instrument that is
correlated with the former but not the latter. The lagged variables in (21) mean that equation is not
estimable for t < 3, so consider when t = 3. We have

yi3 − yi2 = ρ(yi2 − yi1) + (xi3 − xi2)′β + (εi3 − εi2) (22)

http://www.stata.com/manuals13/dreshape.pdf#dreshape
http://www.stata.com/manuals13/xtxtabond.pdf#xtxtabond
http://www.stata.com/manuals13/xtxtdpd.pdf#xtxtdpd
http://www.stata.com/manuals13/xtxtdpdsys.pdf#xtxtdpdsys

In the Arellano–Bond (1991) estimator, lagged levels of the dependent variable are used as instruments.
With our assumption that the εit are i.i.d., (20) intimates that yi1 can serve as an instrumental variable
when we fit (22).

Next consider (21) when t = 4. We have

yi4 − yi3 = ρ(yi3 − yi2) + (xi4 − xi3)′β + (εi4 − εi3)

Now (20) shows that both yi1 and yi2 are uncorrelated with the error term (εi4 − εi3), so we have
two instruments available. For t = 5, you can show that yi1, yi2, and yi3 can serve as instruments.
As may now be apparent, one of the key features of these dynamic panel-data models is that the
available instruments depend on the time period, t, as was the case for some of the panel Poisson
models we considered earlier. Because the xit are strictly exogenous by assumption, they can serve
as their own instruments.

The initial weight matrix that is appropriate for the GMM dynamic panel-data estimator is slightly
more involved than the unadjusted matrix we have used in most of our previous examples that assumes
the errors are i.i.d. First, rewrite (21) for panel i as

yi − yLi = ρ (yLi − yLLi ) + (Xi −XL
i )β + (εi − εLi )

where yi = (yi3, . . . , yiT ) and yLi = (yi2, . . . , yi,T−1), yLLi = (yi1, . . . , yi,T−2), and Xi, XL
i , εi,

and εLi are defined analogously. Let Z denote the full matrix of instruments for panel i, including the
variables specified in both the instruments() and xtinstruments() options; the exact structure
is detailed in Methods and formulas.

By assumption, εit is i.i.d., so the first-difference (εit − εi,t−1) is necessarily autocorrelated
with correlation −0.5. Therefore, we should not use a weight matrix that assumes the errors are
independent. For dynamic panel-data models, we can show that the appropriate initial weight matrix
is

Ŵ =

(
1

N

∑
i
Z′iHDZi

)−1
where

HD =


1 −0.5 0 . . . 0 0
−0.5 1 −0.5 . . . 0 0

...
...

...
. . .

...
...

0 0 0 . . . 1 −0.5
0 0 0 . . . −0.5 1


We can obtain this initial weight matrix by specifying winitial(xt D). The letter D indicates that
the equation we are estimating is specified in first-differences.

Example 15: Arellano–Bond estimator

Say we want to fit the model

nit = ρ ni,t−1 + β1wit + β2wi,t−1 + β3kit + β4ki,t−1 + ui + εit (23)

where we assume that wit and kit are strictly exogenous. First-differencing, our residual equation is

ε∗it = (εit − εi,t−1) =nit − ni,t−1 − ρ (ni,t−1 − ni,t−2)− β1(wit − wi,t−1)

− β2(wi,t−1 − wi,t−2)− β3(kit − ki,t−1)− β4(ki,t−1 − ki,t−2) (24)

In Stata, we type

. use http://www.stata-press.com/data/r13/abdata

. gmm (D.n - {rho}*LD.n - {xb:D.w LD.w D.k LD.k}),
> xtinstruments(n, lags(2/.)) instruments(D.w LD.w D.k LD.k, noconstant)
> deriv(/rho = -1*LD.n) deriv(/xb = -1) winitial(xt D) onestep

Step 1
Iteration 0: GMM criterion Q(b) = .0011455
Iteration 1: GMM criterion Q(b) = .00009103
Iteration 2: GMM criterion Q(b) = .00009103 (backed up)

GMM estimation

Number of parameters = 5
Number of moments = 32
Initial weight matrix: XT D Number of obs = 751

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/rho .8041712 .1199819 6.70 0.000 .5690111 1.039331
/xb_D_w -.5600476 .1619472 -3.46 0.001 -.8774583 -.242637

/xb_LD_w .3946699 .1092229 3.61 0.000 .1805969 .6087429
/xb_D_k .3520286 .0536546 6.56 0.000 .2468676 .4571897

/xb_LD_k -.2160435 .0679689 -3.18 0.001 -.3492601 -.0828269

Instruments for equation 1:
XT-style: L(2/.).n
Standard: D.w LD.w D.k LD.k

Because w and k are strictly exogenous, we specified the variants of them that appear in (24) in the
instruments() option; because there is no constant term in the model, we specified noconstant
to omit the constant from the instrument list.

We specified xtinstruments(n, lags(2/.)) to tell gmm what instruments to use for the lagged
dependent variable included as a regressor in (23). Based on our previous discussion, lags two and
higher of nit can serve as instruments. The lags(2/.) suboption tells gmm that the first available
instrument for nit is the lag-two value ni,t−2. The “.” tells gmm to use all further lags of nit as
instruments as well. The instrument matrices in dynamic panel-data models can become large if the
dataset has many time periods per panel. In those cases, you could specify, for example, lags(2/4)
to use just lags two through four instead of using all available lags.

Our results are identical to those we would obtain using xtabond with the syntax

xtabond n L(0/1).w L(0/1).k, lags(1) noconstant vce(robust)

Had we left off the vce(robust) option in our call to xtabond, we would have had to specify
vce(unadjusted) in our call to gmm to obtain the same standard errors.

Technical note
gmm automatically excludes observations for which there are no valid observations for the panel-

style instruments. However, it keeps in the estimation sample those observations for which fewer than
the maximum number of instruments you requested are available. For example, if you specify the
lags(2/4) suboption, you have requested three instruments, but gmm will keep observations even if
only one or two instruments are available.

Example 16: Two-step Arellano–Bond estimator

Here we refit the model from example 15, using the two-step GMM estimator.

. gmm (D.n - {rho}*LD.n - {xb:D.w LD.w D.k LD.k}),
> xtinstruments(n, lags(2/.)) instruments(D.w LD.w D.k LD.k, noconstant)
> deriv(/rho = -1*LD.n) deriv(/xb = -1) winitial(xt D) wmatrix(robust)
> vce(unadjusted)

Step 1
Iteration 0: GMM criterion Q(b) = .0011455
Iteration 1: GMM criterion Q(b) = .00009103
Iteration 2: GMM criterion Q(b) = .00009103 (backed up)

Step 2
Iteration 0: GMM criterion Q(b) = .44107941
Iteration 1: GMM criterion Q(b) = .4236729
Iteration 2: GMM criterion Q(b) = .4236729

GMM estimation

Number of parameters = 5
Number of moments = 32
Initial weight matrix: XT D Number of obs = 751
GMM weight matrix: Robust

Coef. Std. Err. z P>|z| [95% Conf. Interval]

/rho .8044783 .0534763 15.04 0.000 .6996667 .90929
/xb_D_w -.5154978 .0335506 -15.36 0.000 -.5812557 -.4497399

/xb_LD_w .4059309 .0637294 6.37 0.000 .2810235 .5308384
/xb_D_k .3556204 .0390892 9.10 0.000 .2790071 .4322337

/xb_LD_k -.2204521 .046439 -4.75 0.000 -.3114709 -.1294332

Instruments for equation 1:
XT-style: L(2/.).n
Standard: D.w LD.w D.k LD.k

Our results match those you would obtain with the command

xtabond n L(0/1).(w k), lags(1) noconstant twostep

Technical note
Had we specified vce(robust) in our call to gmm, we would have obtained the traditional

sandwich-based robust covariance matrix, but our standard errors would not match those we would
obtain by specifying vce(robust) with the xtabond command. The xtabond, xtdpd, and xtdpdsys
commands implement a bias-corrected robust VCE for the two-step GMM dynamic panel-data estimator.
Traditional VCEs computed after the two-step dynamic panel-data estimator have been shown to exhibit
often-severe bias; see Windmeijer (2005).

Neither of the two dynamic panel-data examples (15 and 16) we have fit so far include a constant
term. When a constant term is included, the dynamic panel-data estimator is in fact a two-equation
system estimator. For notational simplicity, consider a simple model containing just a constant term
and one lag of the dependent variable:

yit = α+ ρyi,t−1 + ui + εit

First-differencing to remove the ui term, we have

yit − yi,t−1 = ρ(yi,t−1 − yi,t−2) + (εit − εi,t−1) (25)

This has also eliminated the constant term. If we assume E(ui) = 0, which is reasonable if a constant
term is included in the model, then we can recover α by including the moment condition

yit = α+ ρyi,t−1 + ε′it (26)

where ε′it = ui + εit. The parameter ρ continues to be identified by (25), so the only instrument we
use with (26) is a constant term. As before, the error term (εi,t− εi,t−1) is necessarily autocorrelated
with correlation coefficient −0.5, though the error term ε′it is white noise. Therefore, our initial weight
matrix should be

Ŵ =

(
1

N

∑
i
Z′iHZi

)−1
where

H =

[
HD 0
0 I

]
and I is a conformable identity matrix.

One complication arises concerning the relevant estimation sample. Looking at (25), we apparently
lose the first two observations from each panel because of the presence of yi,t−2, but in (26) we need
only to sacrifice one observation, for yi,t−1. For most multiple-equation models, we need to use the
same estimation sample for all equations. However, in dynamic panel-data models, we can use more
observations to fit the equation in level form [(26) here] than the equation in first-differences [equation
(25)]. To request this behavior, we specify the nocommonesample option to gmm. That option tells
gmm to use as many observations as possible for each equation, ignoring the loss of observations due
to lagging or differencing.

Example 17: Arellano–Bond estimator with constant term

Here we fit the model
nit = α+ ρ ni,t−1 + ui + εit

Without specifying derivatives, our command would be

. gmm (D.n - {rho}*LD.n) (n - {alpha} - {rho}*L.n),
> xtinstruments(1: n, lags(2/.)) instruments(1:, noconstant) onestep
> winitial(xt DL) vce(unadj) nocommonesample

We would specify winitial(xt DL) to obtain the required initial weight matrix. The notation DL
indicates that our first moment equation is in first-differences and the second moment equation is
in levels (not first-differenced). We exclude a constant in the instrument list for the first equation,
because first-differencing removed the constant term. Because we do not specify the instruments()
option for the second moment equation, a constant is used by default.

This example also provides us the opportunity to illustrate how to specify derivatives for multiple-
equation GMM models. Within the derivative() option, instead of specifying just the parameter
name, now you must specify the equation name or number, a slash, and the parameter name to which
the derivative applies. In Stata, we type

. gmm (D.n - {rho}*LD.n) (n - {alpha} - {rho}*L.n),
> xtinstruments(1: n, lags(2/.)) instruments(1:, noconstant)
> derivative(1/rho = -1*LD.n) derivative(2/alpha = -1)
> derivative(2/rho = -1*L.n) winitial(xt DL) vce(unadj)
> nocommonesample onestep

Step 1
Iteration 0: GMM criterion Q(b) = .09894466
Iteration 1: GMM criterion Q(b) = .00023508
Iteration 2: GMM criterion Q(b) = .00023508

GMM estimation

Number of parameters = 2
Number of moments = 29
Initial weight matrix: XT DL Number of obs = *

Coef. Std. Err. z P>|z| [95% Conf. Interval]

/rho 1.023349 .0608293 16.82 0.000 .9041259 1.142572
/alpha -.0690864 .0660343 -1.05 0.295 -.1985112 .0603384

* Number of observations for equation 1: 751
Number of observations for equation 2: 891

Instruments for equation 1:
XT-style: L(2/.).n

Instruments for equation 2:
Standard: _cons

These results are identical to those we would obtain by typing

xtabond n, lags(1)

Because we specified nocommonesample, gmm did not report the number of observations used in
the header of the output. In this dataset, there are in fact 1,031 observations on 140 panels. In the
second equation, the presence of the lagged value of n reduces the sample size for that equation to
1031− 140 = 891. In the first equation, we lose the first two observations per panel due to lagging
and differencing, leading to 751 usable observations. These tallies are listed after the coefficient table
in the output.

Technical note
Specifying

xtinstruments(x1 x2 x3, lags(1/3))

differs from
instruments(L(1/3).(x1 x2 x3))

in how observations are excluded from the estimation sample. When you use the latter syntax, gmm
must exclude the first three observations from each panel when computing the moment equation: you
requested three lags of each regressor be used as instruments, so the first residual that could be interacted
with those instruments is the one for t = 4. On the other hand, when you use xtinstruments(), you
are telling gmm that you would like to use up to the first three lags of x1, x2, and x3 as instruments
but that using just one lag is acceptable. Because most panel datasets have a relatively modest number
of observations per panel, dynamic instrument lists are typically used so that the number of usable
observations is maximized. Dynamic instrument lists also accommodate the fact that there are more
valid instruments for later time periods than earlier time periods.

Specifying panel-style instruments using the xtinstruments() option also affects how the standard
instruments specified in the instruments() option are treated. To illustrate, suppose we have a
balanced panel dataset with T = 5 observations per panel and we specify

. gmm . . ., xtinstruments(w, lags(1/2)) instruments(x)

We will lose the first observation because we need at least one lag of w to serve as an instrument.
Our instrument matrix for panel i will therefore be

Zi =



wi1 0 0 0
0 wi1 0 0
0 wi2 0 0
0 0 wi2 0
0 0 wi3 0
0 0 0 wi3
0 0 0 wi4
xi2 xi3 xi4 xi5
1 1 1 1


(27)

The vector of ones in the final row represents the constant term implied by the instruments()
option. Because we lost the first observation, the residual vector ui will be 4× 1. Thus our moment
conditions for the ith panel can be written in matrix notation as

E{Ziui(β)} = E

Zi


ui2(β)
ui3(β)
ui4(β)
ui5(β)


 = 0

The moment conditions corresponding to the final two rows of (27) say that

E

{
T=4∑
t=2

xituit(β)

}
= 0 and E

{
T=4∑
t=2

uit(β)

}
= 0

Because we specified panel-style instruments with the xtinstruments() option, gmm no longer
uses moment conditions for strictly exogenous variables of the form E{xituit(β)} = 0 for each t.
Instead, the moment conditions now stipulate that the average (over t) of xituit(β) has expectation
zero. This corresponds to the approach proposed by Arellano and Bond (1991, 280) and others.

When you request panel-style instruments with the xtinstruments() option, the number of
instruments in the Zi matrix increases quadratically in the number of periods. The dynamic panel-
data estimators we have discussed in this section are designed for datasets that contain a large number
of panels and a modest number of time periods. When the number of time periods is large, estimators
that use standard (non–panel-style) instruments are more appropriate.

We have focused on the Arellano–Bond dynamic panel-data estimator because of its relative
simplicity. gmm can additionally fit any models that can be formulated using the xtdpd and xtdpdsys
commands; see [XT] xtdpd and [XT] xtdpdsys. The key is to determine the appropriate instruments
to use for the level and difference equations. You may find it useful to fit a version of your model
with those commands to determine what instruments and XT-style instruments to use. We conclude
this section with an example using the Arellano–Bover/Blundell–Bond estimator.

http://www.stata.com/manuals13/xtxtdpd.pdf#xtxtdpd
http://www.stata.com/manuals13/xtxtdpdsys.pdf#xtxtdpdsys

Example 18: Arellano–Bover/Blundell–Bond estimator

We fit a small model that includes one lag of the dependent variable n as a regressor as well as
the contemporaneous and first lag of w, which we assume are strictly exogenous. We could fit our
model using xtdpdsys using the syntax

xtdpdsys n L(0/1).w, lags(1) twostep

Applying virtually all the syntax issues we have discussed so far, the equivalent gmm command is

. gmm (n - {rho}*L.n - {w}*w - {lagw}*L.w - {c})
> (D.n - {rho}*LD.n - {w}*D.w - {lagw}*LD.w),
> xtinst(1: D.n, lags(1/1)) xtinst(2: n, lags(2/.))
> inst(2: D.w LD.w, noconstant)
> deriv(1/rho = -1*L.n) deriv(1/w = -1*w)
> deriv(1/lagw = -1*L.w) deriv(1/c = -1)
> deriv(2/rho = -1*LD.n) deriv(2/w = -1*D.w)
> deriv(2/lagw = -1*LD.w)
> winit(xt LD) wmatrix(robust) vce(unadjusted)
> nocommonesample

Step 1
Iteration 0: GMM criterion Q(b) = .10170339
Iteration 1: GMM criterion Q(b) = .00022772
Iteration 2: GMM criterion Q(b) = .00022772

Step 2
Iteration 0: GMM criterion Q(b) = .59965014
Iteration 1: GMM criterion Q(b) = .56578186
Iteration 2: GMM criterion Q(b) = .56578186

GMM estimation

Number of parameters = 4
Number of moments = 39
Initial weight matrix: XT LD Number of obs = *
GMM weight matrix: Robust

Coef. Std. Err. z P>|z| [95% Conf. Interval]

/rho 1.122738 .0206512 54.37 0.000 1.082263 1.163214
/w -.6719909 .0246148 -27.30 0.000 -.7202351 -.6237468

/lagw .571274 .0403243 14.17 0.000 .4922398 .6503083
/c .154309 .17241 0.90 0.371 -.1836084 .4922263

* Number of observations for equation 1: 891
Number of observations for equation 2: 751

Instruments for equation 1:
XT-style: LD.n
Standard: _cons

Instruments for equation 2:
XT-style: L(2/.).n
Standard: D.w LD.w

Details of moment-evaluator programs

In examples 9, 10, 11, and 12, we used moment-evaluator programs to evaluate moment conditions
that could not be specified using the interactive version of gmm. In example 11, we also showed how to
pass additional information to an evaluator program. Here we discuss how to make moment-evaluator
programs provide derivatives and accept weights.

The complete specification for a moment-evaluator program’s syntax statement is

syntax varlist if [weight], at(name) options [derivatives(varlist)]

The macro ‘varlist’ contains the list of variables that we are to fill in with the values of our
residual equations. The macro ‘if’ represents an if condition that restricts the estimation sample.
The macro ‘at’ represents a vector containing the parameter values at which we are to evaluate our
residual equations. options represent other options that you specify in your call to gmm and want to
have passed to your moment-evaluator programs. In example 11, we included the mylhs() option
so that we could pass the name of the dependent variable to our evaluator program.

Two new elements of the syntax statement allow for weights and derivatives. weight specifies
the types of weights your program allows. The interactive version of gmm allows for fweights,
aweights, and pweights. However, unless you explicitly allow your moment evaluator program to
accept weights, you cannot specify weights in your call to gmm with the moment-evaluator program
version.

The derivatives() option is used to pass to your program a set of variables that you are to fill
in with the derivatives of your residual equations with respect to the parameters.

To indicate that your program can calculate derivatives, you specify either the hasderivatives
or the haslfderivatives option to gmm. The hasderivatives option indicates that your program
calculates parameter-level derivatives; that method requires more work but can be applied to any GMM
problem. The haslfderivatives option requires less work but can be used only when the model’s
residual equations satisfy certain restrictions and you use the <eqname>:<varname> syntax with
the parameters() option.

We first consider how to write the derivative computation logic to work with the hasderivatives
option and provide an example; then we do the same for the haslfderivatives option.

Say you specify k parameters in the nparameters() or parameters() option and q equations in the
nequations() or equations() option and you specify hasderivatives. Then ‘derivatives’
will contain k× q variables. The first k variables are for the derivatives of the first residual equation
with respect to the k parameters, the second k variables are for the derivatives of the second residual
equation, and so on.

Example 19: Specifying derivatives with simple parameter names

To focus on how to specify derivatives, we return to the simple moment-evaluator program we
used in example 9, in which we had three regressors, and extend it to supply derivatives. The residual
equation corresponding to moment condition (13) is

uit(β) = yit − µit
yi
µi

where µit, µi, and yi were defined previously. Now

∂

∂βj
uit(β) = −µit

yi
µ2
i

(
x
(j)
it µi −

1

T

l=T∑
l=1

x
(j)
il µil

)
(28)

where x(j)it represents the jth element of xit.

Our moment-evaluator program is

program gmm_poideriv

version 13
syntax varlist if, at(name) [derivatives(varlist)]

quietly {

// Calculate residuals as before
tempvar mu mubar ybar

gen double ‘mu’ = exp(x1*‘at’[1,1] + x2*‘at’[1,2] ///
+ x3*‘at’[1,3]) ‘if’

egen double ‘mubar’ = mean(‘mu’) ‘if’, by(id)
egen double ‘ybar’ = mean(y) ‘if’, by(id)

replace ‘varlist’ = y - ‘mu’*‘ybar’/‘mubar’ ‘if’

// Did -gmm- request derivatives?
if "‘derivatives’" == "" {

exit // no, so we are done
}
// Calculate derivatives
// We need the panel means of x1*mu, x2*mu, and x3*mu
tempvar work x1mubar x2mubar x3mubar
generate double ‘work’ = x1*‘mu’ ‘if’
egen double ‘x1mubar’ = mean(‘work’) ‘if’, by(id)

replace ‘work’ = x2*‘mu’ ‘if’
egen double ‘x2mubar’ = mean(‘work’) ‘if’, by(id)

replace ‘work’ = x3*‘mu’ ‘if’
egen double ‘x3mubar’ = mean(‘work’) ‘if’, by(id)

local d1: word 1 of ‘derivatives’
local d2: word 2 of ‘derivatives’
local d3: word 3 of ‘derivatives’
replace ‘d1’ = -1*‘mu’*‘ybar’/‘mubar’^2*(x1*‘mubar’ - ‘x1mubar’)
replace ‘d2’ = -1*‘mu’*‘ybar’/‘mubar’^2*(x2*‘mubar’ - ‘x2mubar’)
replace ‘d3’ = -1*‘mu’*‘ybar’/‘mubar’^2*(x3*‘mubar’ - ‘x3mubar’)

}

end

The derivatives() option is made optional in the syntax statement by placing it in square brackets.
If gmm needs to evaluate your moment equations but does not need derivatives at that time, then the
derivatives() option will be empty. In our program, we check to see if that is the case, and, if
so, exit without calculating derivatives. As is often the case with [R] ml as well, the portion of our
program devoted to derivatives is longer than the code to compute the objective function.

The first part of our derivative code computes the term

1

T

l=T∑
l=1

x
(j)
il µil (29)

for x(j)it = x1, x2, and, x3. The ‘derivatives’ macro contains three variable names, corresponding
to the three parameters of the ‘at’ matrix. We extract those names into local macros ‘d1’, ‘d2’,
and ‘d3’, and then fill in the variables those macros represent with the derivatives shown in (28).

http://www.stata.com/manuals13/rml.pdf#rml

With our program written, we fit our model by typing

. use http://www.stata-press.com/data/r13/poisson1

. gmm gmm_poideriv, nequations(1) parameters(b1 b2 b3)
> instruments(x1 x2 x3, noconstant) vce(cluster id) onestep hasderivatives

Step 1
Iteration 0: GMM criterion Q(b) = 51.99142
Iteration 1: GMM criterion Q(b) = .04345191
Iteration 2: GMM criterion Q(b) = 8.720e-06
Iteration 3: GMM criterion Q(b) = 7.115e-13
Iteration 4: GMM criterion Q(b) = 5.129e-27

GMM estimation

Number of parameters = 3
Number of moments = 3
Initial weight matrix: Unadjusted Number of obs = 409

(Std. Err. adjusted for 45 clusters in id)

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/b1 1.94866 .1000265 19.48 0.000 1.752612 2.144709
/b2 -2.966119 .0923592 -32.12 0.000 -3.14714 -2.785099
/b3 1.008634 .1156561 8.72 0.000 .781952 1.235315

Instruments for equation 1: x1 x2 x3

Our results are identical to those in example 9. Another way to verify that our program calculates
derivatives correctly would be to type

. gmm gmm_poideriv, nequations(1) parameters(b1 b2 b3)
> instruments(x1 x2 x3, noconstant) vce(cluster id) onestep

Without the hasderivatives or haslfderivatives option, gmm will not request derivatives from
your program, even if it contains code to compute them. If you have trouble obtaining convergence
with the hasderivatives or haslfderivatives option but do not have trouble without specifying
one of them, then you need to recheck your derivatives.

After example 9, we remarked that the evaluator program would have to be changed to ac-
commodate different regressors. We then showed how you can specify parameters using the syntax
<eqname>:<varname> and then use matrix score to compute linear combinations of variables.
To specify derivatives when you specify parameters using this equation-name syntax, ensure that your
residual equations satisfy the “linear-form restriction” analogous to the restrictions of linear-form
evaluators used by ml. See [R] ml and Gould, Pitblado, and Poi (2010) for more information about
linear-form evaluators.

A GMM residual equation satisfies the linear-form restriction if the equation can be written in
terms of a single observation in the dataset and if the equation for observation i does not depend on
any observations j 6= i. Cross-sectional models satisfy the linear-form restriction. Time-series models
satisfy the linear-form restriction only when no lags or leads are used.

Panel-data models often do not satisfy the linear-form restriction. For example, recall moment
condition (13) for a panel Poisson model. That residual equation included panel-level mean terms ȳi
and µ̄i, so the residual equation for an individual observation depends on all the observations in the
same panel.

When a residual equation does not satisfy the linear-form restriction, neither will its derivatives. To
apply the chain rule, we need a way to multiply the eqname-level derivative by each of the variables

http://www.stata.com/manuals13/rml.pdf#rml

in the equation to obtain parameter-level derivatives. In (28), for example, there is no way to factor
out each x

(j)
it variable and obtain an eqname-level derivative that we then multiply by each of the

x
(j)
it s.

Suppose we do have a model with q = 2 moment equations, both of which do satisfy the linear-form
restriction, and we specify the parameters() option like this:

. gmm . . ., parameters(eq1:x1 eq1:x2 eq1:_cons eq2:_cons eq3:x1 eq3:x2 eq3:_cons)

We have specified n = 3 eqnames in the parameters() option: eq1, eq2, and eq3. When we specify
the haslfderivatives option, gmm will pass n× q = 3× 2 = 6 variables in the derivatives()
option. The first three variables are to be filled with

∂

∂eq1
u1i(β)

∂

∂eq2
u1i(β) and

∂

∂eq3
u1i(β)

where u1i(β) is the ith observation for the first moment equation. Then the second three variables
are to be filled with

∂

∂eq1
u2i(β)

∂

∂eq2
u2i(β) and

∂

∂eq3
u2i(β)

where u2i(β) is the second moment equation. In this example, we filled in a total of six variables with
derivatives. If we instead used the hasderivatives option, we would have filled k×q = 7×2 = 14
variables; moreover, if we wanted to change the number of variables in our model, we would have
modified our evaluator program.

Example 20: Specifying derivatives with linear-form residual equations

In examples 7 and 8, we showed how to specify derivatives with an exponential regression model
when using the interactive version of gmm. Here we show how to write a moment-evaluator program
for the exponential regression model, including derivatives.

The residual equation for observation i is

ui = yi − exp(x′iβ)

where xi may include a constant term. The derivative with respect to the linear combination x′iβ is

∂ui
∂x′iβ

= − exp(x′iβ)

To verify this residual equation satisfies the linear-form restriction, we see that for the jth element
of β, we have

∂ui
∂βj

= −xij exp(x′iβ) =
∂ui
∂x′iβ

× xij

so that given ∂ui/∂x′iβ, gmm can apply the chain rule to obtain the derivatives with respect to the
individual parameters.

Our moment-evaluator program is
program gmm_poideriv2

version 13
syntax varlist if, at(name) [derivatives(varlist)]

quietly {
tempvar mu
matrix score double ‘mu’ = ‘at’ ‘if’, eq(#1)
replace ‘mu’ = exp(‘mu’)

local depvar : coleq ‘at’
local depvar : word 1 of ‘depvar’
replace ‘varlist’ = ‘depvar’ - ‘mu’ ‘if’

// Did -gmm- request derivatives?
if "‘derivatives’" == "" {

exit // no, so we are done
}
// Calculate derivatives
// The derivatives macro only has one variable
// for this model.

replace ‘derivatives’ = -1*‘mu’ ‘if’
}

end

To fit our model of doctor visits treating income as an endogenous regressor, we type
. use http://www.stata-press.com/data/r13/docvisits

. gmm gmm_poideriv2, nequations(1)
> instruments(private chronic female age black hispanic)
> parameters(docvis:private docvis:chronic
> docvis:female docvis:income docvis:_cons) haslfderivatives

Step 1
Iteration 0: GMM criterion Q(b) = 16.910173
Iteration 1: GMM criterion Q(b) = .82270871
Iteration 2: GMM criterion Q(b) = .21831995
Iteration 3: GMM criterion Q(b) = .12685934
Iteration 4: GMM criterion Q(b) = .12672369
Iteration 5: GMM criterion Q(b) = .12672365

Step 2
Iteration 0: GMM criterion Q(b) = .00234641
Iteration 1: GMM criterion Q(b) = .00215957
Iteration 2: GMM criterion Q(b) = .00215911
Iteration 3: GMM criterion Q(b) = .00215911

GMM estimation

Number of parameters = 5
Number of moments = 7
Initial weight matrix: Unadjusted Number of obs = 4412
GMM weight matrix: Robust

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

private .535335 .159904 3.35 0.001 .221929 .848741
chronic 1.090126 .0617659 17.65 0.000 .9690668 1.211185
female .6636579 .0959885 6.91 0.000 .475524 .8517918
income .0142855 .0027162 5.26 0.000 .0089618 .0196092
_cons -.5983477 .138433 -4.32 0.000 -.8696714 -.327024

Instruments for equation 1: private chronic female age black hispanic _cons

Our results match those shown in example 8.

We can change the variables in our model just by changing the parameters() and instruments()
options; we do not need to make any changes to the moment-evaluator program, because we used
linear-form derivatives.

Depending on your model, allowing your moment-evaluator program to accept weights may be as
easy as modifying the syntax command to allow them, or it may require significantly more work.
If your program uses only commands like generate and replace, then just modifying the syntax
command is all you need to do; gmm takes care of applying the weights to the observation-level residuals
when computing the sample moments, derivatives, and weight matrices. On the other hand, if your
moment-evaluator program computes residuals using statistics that depend on multiple observations,
then you must apply the weights passed to your program when computing those statistics.

In our examples of panel Poisson with strictly exogenous regressors (9 and 18), we used the
statistics µi and yi when computing the residuals. If we are to allow weights with our moment-
evaluator program, then we must incorporate those weights when computing µi and yi. Moreover,
looking at the derivative in (28), the term highlighted in (29) is in fact a sample mean, so we must
incorporate weights when computing it.

Example 21: Panel Poisson with derivatives and weights

Here we modify the program in example 19. to accept frequency weights. One complication
immediately arises: we had been using egen to compute µi and yi. egen does not accept weights, so
we must compute µi and yi ourselves, incorporating any weights the user may specify. Our program
is

program gmm_poiderivfw

version 13
syntax varlist if [fweight/], at(name) [derivatives(varlist)]

quietly {
if "‘exp’" == "" { // no weights

local exp 1 // weight each observation equally
}

// Calculate residuals as before
tempvar mu mubar ybar sumwt

gen double ‘mu’ = exp(x1*‘at’[1,1] + x2*‘at’[1,2] ///
+ x3*‘at’[1,3]) ‘if’

bysort id: gen double ‘sumwt’ = sum(‘exp’)
by id: gen double ‘mubar’ = sum(‘mu’*‘exp’)
by id: gen double ‘ybar’ = sum(y*‘exp’)
by id: replace ‘mubar’ = ‘mubar’[_N] / ‘sumwt’[_N]
by id: replace ‘ybar’ = ‘ybar’[_N] / ‘sumwt’[_N]

replace ‘varlist’ = y - ‘mu’*‘ybar’/‘mubar’ ‘if’

// Did -gmm- request derivatives?
if "‘derivatives’" == "" {

exit // no, so we are done
}
// Calculate derivatives
// We need the panel means of x1*mu, x2*mu, and x3*mu
tempvar work x1mubar x2mubar x3mubar
generate double ‘work’ = x1*‘mu’ ‘if’
by id: generate double ‘x1mubar’ = sum(‘work’*‘exp’)
by id: replace ‘x1mubar’ = ‘x1mubar’[_N] / ‘sumwt’[_N]

replace ‘work’ = x2*‘mu’ ‘if’
by id: generate double ‘x2mubar’ = sum(‘work’*‘exp’)
by id: replace ‘x2mubar’ = ‘x2mubar’[_N] / ‘sumwt’[_N]

replace ‘work’ = x3*‘mu’ ‘if’
by id: generate double ‘x3mubar’ = sum(‘work’*‘exp’)
by id: replace ‘x3mubar’ = ‘x3mubar’[_N] / ‘sumwt’[_N]

local d1: word 1 of ‘derivatives’
local d2: word 2 of ‘derivatives’
local d3: word 3 of ‘derivatives’
replace ‘d1’ = -1*‘mu’*‘ybar’/‘mubar’^2*(x1*‘mubar’ - ‘x1mubar’)
replace ‘d2’ = -1*‘mu’*‘ybar’/‘mubar’^2*(x2*‘mubar’ - ‘x2mubar’)
replace ‘d3’ = -1*‘mu’*‘ybar’/‘mubar’^2*(x3*‘mubar’ - ‘x3mubar’)

}

end

Our syntax command now indicates that fweights are allowed. The first part of our code looks
at the macro ‘exp’. If it is empty, then the user did not specify weights in their call to gmm; and
we set the macro equal to 1, so that we weight each observation equally. After we compute µit, we
calculate µi and yi, taking into account weights. To compute frequency-weighted means for each
panel, we just multiply each observation by its respective weight, sum over all observations in the
panel, then divide by the sum of the weights for the panel. (See [U] 20.23 Weighted estimation for
information on how to handle aweights and pweights.) We use the same procedure to compute the
frequency-weighted variant of expression (29) in the derivative calculations.

http://www.stata.com/manuals13/u20.pdf#u20.23Weightedestimation

To use our program, we type

. use http://www.stata-press.com/data/r13/poissonwts

. gmm gmm_poiderivfw [fw=fwt], nequations(1) parameters(b1 b2 b3)
> instruments(x1 x2 x3, noconstant) vce(cluster id) onestep hasderivatives
(sum of wgt is 819)

Step 1
Iteration 0: GMM criterion Q(b) = 49.8292
Iteration 1: GMM criterion Q(b) = .11136736
Iteration 2: GMM criterion Q(b) = .00008519
Iteration 3: GMM criterion Q(b) = 7.110e-11
Iteration 4: GMM criterion Q(b) = 5.596e-23

GMM estimation

Number of parameters = 3
Number of moments = 3
Initial weight matrix: Unadjusted Number of obs = 819

(Std. Err. adjusted for 45 clusters in id)

Robust
Coef. Std. Err. z P>|z| [95% Conf. Interval]

/b1 1.967766 .111795 17.60 0.000 1.748652 2.186881
/b2 -3.060838 .0935561 -32.72 0.000 -3.244205 -2.877472
/b3 1.037594 .1184227 8.76 0.000 .80549 1.269698

Instruments for equation 1: x1 x2 x3

Testing whether our program works correctly with frequency weights is easy. A frequency-weighted
dataset is just a compact form of a larger dataset in which identical observations are omitted and a
frequency-weight variable is included to tell us how many times each observation in the smaller dataset
appears in the larger dataset. Therefore, we can expand our smaller dataset by the frequency-weight
variable and then refit our model without specifying frequency weights. If we obtain the same results,
our program works correctly. When we type

. expand fw

. gmm gmm_poiderivfw, nequations(1) parameters(b1 b2 b3)
> instruments(x1 x2 x3, noconstant) vce(cluster id) onestep

we obtain the same results as before.

Stored results
gmm stores the following in e():

Scalars
e(N) number of observations
e(k) number of parameters
e(k eq) number of equations in e(b)
e(k eq model) number of equations in overall model test
e(k aux) number of auxiliary parameters
e(n moments) number of moments
e(n eq) number of equations in moment-evaluator program
e(Q) criterion function
e(J) Hansen J χ2 statistic
e(J df) J statistic degrees of freedom
e(k i) number of parameters in equation i

e(has xtinst) 1 if panel-style instruments specified, 0 otherwise
e(N clust) number of clusters
e(type) 1 if interactive version, 2 if moment-evaluator program version
e(rank) rank of e(V)
e(ic) number of iterations used by iterative GMM estimator
e(converged) 1 if converged, 0 otherwise

Macros
e(cmd) gmm
e(cmdline) command as typed
e(title) title specified in title()
e(title 2) title specified in title2()
e(clustvar) name of cluster variable
e(inst i) equation i instruments
e(eqnames) equation names
e(winit) initial weight matrix used
e(winitname) name of user-supplied initial weight matrix
e(estimator) onestep, twostep, or igmm
e(rhs) variables specified in variables()
e(params i) equation i parameters
e(wmatrix) wmtype specified in wmatrix()
e(vce) vcetype specified in vce()
e(vcetype) title used to label Std. Err.
e(params) parameter names
e(sexp i) substitutable expression for equation i

e(evalprog) moment-evaluator program
e(evalopts) options passed to moment-evaluator program
e(nocommonesample) nocommonesample, if specified
e(technique) optimization technique
e(properties) b V
e(estat cmd) program used to implement estat
e(predict) program used to implement predict
e(marginsnotok) predictions disallowed by margins

Matrices
e(b) coefficient vector
e(init) initial values of the estimators
e(Wuser) user-supplied initial weight matrix
e(W) weight matrix used for final round of estimation
e(S) moment covariance matrix used in robust VCE computations
e(N byequation) number of observations per equation, if nocommonesample specified
e(V) variance–covariance matrix
e(V modelbased) model-based variance

Functions
e(sample) marks estimation sample

Methods and formulas
Let q denote the number of moment equations. For observation i, i = 1, . . . , N , write the jth

moment equation as zijuij(βj) for j = 1, . . . , q. zij is a 1×mj vector, where mj is the number
of instruments specified for equation j. Let m = m1 + · · ·+mq .

Our notation can incorporate moment conditions of the form hij(wij ;βj) with instruments wij

by defining zij = 1 and uij(βj) = hij(wij ;βj), so except when necessary we do not distinguish
between the two types of moment conditions. We could instead use notation so that all our moment
conditions are of the form hij(wij ;βj), or we could adopt notation that explicitly combines both
forms of moment equations. However, because moment conditions of the form z′ijuij(βj) are arguably
more common, we use that notation.

Let β denote a k× 1 vector of parameters, consisting of all the unique parameters of β1, . . . ,βq .
Then we can stack the moment conditions and write them more compactly as Z′iui(β), where

Zi =


zi1 0 · · · 0
0 zi2 · · · 0
...

...
. . .

...
0 0 · · · ziq

 and ui(β) =


ui1(β1)
ui2(β2)

...
uiq(βj)


The GMM estimator β̂ is the value of β that minimizes

Q(β) =

{
N−1

N∑
i=1

Z′iui(β)

}′
W

{
N−1

N∑
i=1

Z′iui(β)

}
(A1)

for q × q weight matrix W.

By default, gmm minimizes (A1) using the Gauss–Newton method. See Hayashi (2000, 498) for
a derivation. This technique is typically faster than quasi-Newton methods and does not require
second-order derivatives.

Methods and formulas are presented under the following headings:

Initial weight matrix
Weight matrix
Variance–covariance matrix
Hansen’s J statistic
Panel-style instruments

Initial weight matrix

If you specify winitial(identity), then we set W = Iq .

If you specify winitial(unadjusted), then we create matrix Λ with typical submatrix

Λrs = N−1
N∑
i=1

z′irzis

for r = 1, . . . , q and s = 1, . . . , q. If you include the independent suboption, then we set Λrs = 0
for r 6= s. The weight matrix W equals Λ−1.

If you specify winitial(matname), then we set W equal to Stata matrix matname.

If you specify winitial(xt xtspec), then you must specify one or two items in xtspec, one
for each equation. gmm allows you to specify at most two moment equations when you specify
winitial(xt xtspec), one in first-differences and one in levels. We create the block-diagonal matrix
H with typical block Hj . If the jth element of xtspec is “L”, then Hj is the identity matrix of
suitable dimension. If the jth element of xtspec is “D”, then

Hj =


1 −0.5 0 . . . 0 0
−0.5 1 −0.5 . . . 0 0

...
...

...
. . .

...
...

0 0 0 . . . 1 −0.5
0 0 0 . . . −0.5 1


Then

ΛH = N−1g

g=NG∑
g=1

Z′gHZg

where g indexes panels in the dataset, NG is the number of panels, Zg is the full instrument matrix
for panel g, and W = Λ−1H . See Panel-style instruments below for a discussion of how Zg is formed.

Weight matrix

Specification of the weight matrix applies only to the two-step and iterative estimators. When you
use the onestep option, the wmatrix() option is ignored.

We first evaluate (A1) using the initial weight matrix described above and then compute ui(β̂).
In all cases, W = Λ−1. If you specify wmatrix(unadjusted), then we create Λ to have typical
submatrix

Λrs = σrsN
−1

N∑
i=1

z′irzis

where

σrs = N−1
N∑
i=1

uir(β̂)uis(β̂)

and r and s index moment equations. For all types of weight matrices, if the independent suboption
is specified, then Λrs = 0 for r 6= s, where Λrs measures the covariance between moment conditions
for equations r and s.

If you specify wmatrix(robust), then

Λ = N−1
N∑
i=1

Ziui(β̂)u′i(β̂)Z′i

If you specify wmatrix(cluster clustvar), then

Λ = N−1
c=NC∑
c=1

qcq
′
c



gmm — Generalized method of moments estimation 57

where c indexes clusters, NC is the number of clusters, and

qc =
∑
i∈cj

Ziui(β̂)

If you specify wmatrix(hac kernel
[

#
]
), then

Λ =N−1
N∑
i=1

Ziui(β̂)ui(β̂)′Z′i +

N−1
l=n−1∑
l=1

N∑
i=l+1

K(l,m)
{

Ziui(β̂)u′i−l(β̂)Z′i−l + Zi−lui−l(β̂)u′i(β̂)Z′i

}
where m = # if # is specified and m = N − 2 otherwise. Define z = l/(m + 1). If kernel is
bartlett or nwest, then

K(l,m) =
{

1− z 0 ≤ z ≤ 1
0 otherwise

If kernel is parzen or gallant, then

K(l,m) =

{
1− 6z2 + 6z3 0 ≤ z ≤ 0.5
2(1− z)3 0.5 < z ≤ 1
0 otherwise

If kernel is quadraticspectral or andrews, then

K(l,m) =

{
1 z = 0
3{sin(θ)/θ − cos(θ)}/θ2 otherwise

where θ = 6πz/5.

If wmatrix(hac kernel opt) is specified, then gmm uses Newey and West’s (1994) automatic
lag-selection algorithm, which proceeds as follows. Define h to be an m × 1 vector of ones. Note
that this definition of h is slightly different than the one used by ivregress. There, the element of
h corresponding to the constant term equals zero, effectively ignoring the effect of the constant in
determining the optimal lag length. Here we include the effect of the constant term. Now define

fi = {Z′iui(β)}′ h

σ̂j = N−1
N∑

i=j+1

fifi−j j = 0, . . . ,m∗

ŝ(q) = 2

j=m∗∑
j=1

σ̂jj
q

ŝ(0) = σ̂0 + 2

j=m∗∑
j=1

σ̂j

γ̂ = cγ

{(
ŝ(q)

ŝ(0)

)2
}1/(2q+1)

m = γ̂N1/(2q+1)

where q, m∗, and cγ depend on the kernel specified:

Kernel q m∗ cγ

Bartlett/Newey–West 1 int
{

20(T/100)2/9
}

1.1447
Parzen/Gallant 2 int

{
20(T/100)4/25

}
2.6614

Quadratic spectral/Andres 2 int
{

20(T/100)2/25
}

1.3221

where int(x) denotes the integer obtained by truncating x toward zero. For the Bartlett and Parzen
kernels, the optimal lag is min{int(m),m∗}. For the quadratic spectral kernel, the optimal lag is
min{m,m∗}.

Variance–covariance matrix

If you specify vce(unadjusted), then the VCE matrix is computed as

Var(β̂) = N−1
{

G(β̂)′WG(β̂)
}−1

(A2)

where

G(β̂) = N−1
N∑
i=1

Z′i
∂ui(β)

∂β′

∣∣∣∣
β=β̂

(A3)

For the two-step and iterated estimators, we use the weight matrix W that was used to compute the
final-round estimate β̂.

When you do not specify analytic derivatives, gmm must compute the Jacobian matrix (A3)
numerically. By default, gmm computes each element of the matrix individually by using the Mata
deriv() function; see [M-5] deriv( ). This procedure results in accurate derivatives but can be slow
if your model has many instruments or parameters.

When you specify the quickderivatives option, gmm computes all derivatives corresponding
to parameter βj , j = 1, . . . , q, at once, using two-sided derivatives with a step size of |βj |ε1/3,
where ε is the machine precision of a double precision number (approximately 2.22045 × 10−16).
This method requires just two evaluations of the model’s moments to compute an entire column of
(A3) and therefore has the most impact when you specify many instruments or moment equations so
that (A3) has many rows.

For the one-step estimator, how the unadjusted VCE is computed depends on the type of initial weight
matrix requested and the form of the moment equations. If you specify two or more moment equations
of the form hij(wij ;βj), then gmm issues a warning message and computes a heteroskedasticity-
robust VCE because here the matrix Z′Z is necessarily singular; moreover, here you must use the
identity matrix as the initial weight matrix. Otherwise, if you specify winitial(identity) or
winitial(unadjusted), then gmm first computes an unadjusted weight matrix based on β̂ before
evaluating (A2). If you specify winitial(matname), then (A2) is evaluated based on matname; the
user is responsible for verifying that the VCE and other statistics so produced are appropriate.

All types of robust VCEs computed by gmm take the form

Var(β̂) = N−1
{

G(β̂)′WG(β̂)
}−1

G(β̂)′WSWG(β̂)
{

G(β̂)′WG(β̂)
}−1

http://www.stata.com/manuals13/m-5deriv.pdf#m-5deriv()

For the one-step estimator, W represents the initial weight matrix requested using the winitial()
option, and S is computed based on the specification of the vce() option. The formulas for the S
matrix are identical to the ones that define the Λ matrix in Weight matrix above, except that S is
computed after the moment equations are reevaluated using the final estimate of β̂. For the two-step
and iterated GMM estimators, computation of W is controlled by the wmatrix() option based on
the penultimate estimate of β̂.

For details on computation of the VCE matrix with dynamic panel-data models, see Panel-style
instruments below.

Hansen’s J statistic

Hansen’s (1982) J test of overidentifying restrictions is J = N × Q(β̂). J ∼ χ2(m − k). If
m < k, gmm issues an error message without estimating the parameters. If m = k, the model is
just-identified and J is saved as missing (“.”). For the two-step and iterated GMM estimators, the
J statistic is based on the last-computed weight matrix as determined by the wmatrix() option.
For the one-step estimator, gmm recomputes a weight matrix as described in the second paragraph
of Variance–covariance matrix above. To obtain Hansen’s J statistic, you use estat overid; see
[R] gmm postestimation.

Panel-style instruments

Here we discuss several issues that arise only when you specify panel-style instruments by using
the xtinstruments() option. When you specify the xtinstruments() option, we can no longer
consider the instruments for one observation in isolation; instead, we must consider the instrument
matrix for an entire panel at once. In the following discussion, we let T denote the number of
time periods in a panel. To accommodate unbalanced datasets, conceptually we simply use zeros as
instruments and residuals for time periods that are missing in a panel.

We consider the case where you specify both an equation in levels and an equation in differences,
yielding two residual equations. Let uLpt(β) denote the residual for the level equation for panel p in
period t, and let uDpt(β) denote the residual for the corresponding difference equation. Now define
the 2T × 1 vector up(β) as

up(β) = [uLp1(β), uLp2(β), . . . , uLpT (β), uDp2(β), uDp3(β), . . . , uDpT (β)]

The T + 1 element of up is uDp2(β) since we lose the first observation of the difference equation
because of differencing.

We write the moment conditions for the pth panel as Zpup(β). To see how Zp is defined, let
wL
pt and wD

pt denote the vectors of panel-style instruments for the level and difference equations,
respectively, and let time be denoted by t; we discuss their dimensions momentarily. Also let xLpt
and xDpt denote the vectors of instruments specified in instruments() for the level and difference
equations at time t. Without loss of generality, for our discussion we assume that you specify the
level equation first. Then Zp has the form

http://www.stata.com/manuals13/rgmmpostestimation.pdf#rgmmpostestimation

Zp =



wL
1 0 · · · 0 0 0 · · · 0

0 wL
2 · · · 0 0 0 · · · 0

...
...

. . .
...

...
...

. . .
...

0 0 · · · wL
T 0 0 · · · 0

xL1 xL2 · · · xLT 0 0 · · · 0
0 0 · · · 0 wD

1 0 · · · 0
0 0 · · · 0 0 wD

2 · · · 0
...

...
. . .

...
...

...
. . .

...
0 0 · · · 0 0 0 · · · wD

T

0 0 · · · 0 xD1 xD2 · · · xDT


(A4)

To see how the w vectors are formed, suppose you specify

xtinstruments(eq(1): d, lags( a/ b))

Then wL
t will be a (b− a+ 1)× 1 vector consisting of dt−a, . . . , dt−b. If (t− a) ≤ 0, then instead

we set wL
t = 0. If (t− a) > 0 but (t− b) ≤ 0, then we create wL

t to consist of dt−a, . . . , d1. With
this definition, (b − a + 1) defines the maximum number of lags of d used, but gmm will proceed
with fewer lags if all (b− a+ 1) lags are not available. If you specify two panel-style instruments,
d and e, say, then wL

t will consist of dt−a, . . . , dt−b, et−a, . . . , et−b. wD
t is handled analogously.

The xLt vectors are simply j × 1 vectors, where j is the number of regular instruments specified
with the instruments() option; these vectors include a “1” unless you specify the noconstant
suboption.

Looking carefully at (A4), you will notice that for dynamic panel-data models, moment conditions
corresponding to the instruments xLpt take the form

E

[
t=T∑
t=1

xLptu
L
pt(β)

]
= 0

and likewise for xDpt. Instead of having separate moment conditions for each time period, there is one
moment condition equal to the average of individual periods’ moments. See Arellano and Bond (1991,
280). To include separate moment conditions for each time period, instead of specifying, say,

instruments(1: x)

you could instead first generate a variable called one equal to unity for all observations and specify

xtinstruments(1: x one)

(Creating the variable one is necessary because a constant is not automatically included in variable
lists specified in xtinstruments().)

Unbalanced panels are essentially handled by including zeros rows and columns of Zp and up(β)
corresponding to missing time periods. However, the numbers of instruments and moment conditions
reported by gmm do not reflect this trickery and instead reflect the numbers of instruments and moment
conditions that are not manipulated in this way. Moreover, gmm includes code to work through these
situations efficiently without actually having to fill in zeros.

When you specify winitial(xt . . .), the one-step unadjusted VCE is computed as

Var(β̂) = σ̂2
1ΛH

where ΛH was defined previously,

σ̂2
1 = (N − k)−1

p=P∑
p=1

uDp (β̂)

and uDp (β̂) = [uDp2(β̂), . . . , uDpT (β̂)]. Here we use (N − k)−1 instead of N−1 to match xtdpd.

